

==★== **THE** ==★==

AMSTRAD

PROGRAM

BOOK

PETER GOODE

THE
AMSTRAD
PROGRAM BOOK

**THE
AMSTRAD
PROGRAM BOOK**

PETER GOODE

Phoenix Publishing Associates Ltd
Bushey, Herts.

Copyright © Peter Goode 1984
All rights reserved

First published in Great Britain in 1984 by
PHOENIX PUBLISHING ASSOCIATES LTD
14 Vernon Road, Bushey, Herts. WD2 2JL

ISBN 0 9465 7624 6

Printed in Great Britain by
Billing & Sons Limited, Worcester
Cover design by
Ivor Claydon
Typesetting by
Prestige Press (UK) Ltd.
Chesham, Bucks.
Illustrations by
Denis Gibney
Chorleywood, Herts.

PREFACE

If you are an Amstrad owner, or are about to become one, it is probably because you have decided that this computer offers you more potential for gaming, programming etc. in colour, with sound, than other machines on the market.

Because we are aware of the potential of the Amstrad we have written a wide choice of program which make use of the computer's facilities.

If you are a games player, you will find ample opportunity to use the colour and graphics whilst testing your reactions and finger twisting skills to the full. If you are interested in adventure gaming we have included a program which is a variant on Dragons and Dungeons. We have not, for reasons of space, been able to list a fully extended program but, when you buy your disks, you will be able to use a wider range of circumstances for your adventuring.

To help users understand the programs we have included full 'how to' instructions and, for those who are keen to move on to greater things, programming hints where we feel they may be of interest.

CONTENTS

Important notes	8
1 Zombie Island	9
2 Clear Path	13
3 Castle Adventure	17
4 Minefield	28
5 Asteroid Storm	32
6 Wordgrams	35
7 Driving Test	38
8 Space Trek	41
9 Dragon Dungeon	49
10 Sniper	59
11 Warplan	62
12 S.I.M.P.E.R.	68
13 Wordsquare	74
14 Fast Typer	79
15 Cubist Conundrums	82
16 Wipe Off	84
17 First One	87
18 Weightlifter	90
19 Kaleidoscope	95
20 Zombie Maze	97
21 Pontoon	102
22 Morse Encoder	107
23 Safecracker	110

24 Biorythm Graphing and Prediction	114
25 Dice	118
26 Hunt the Day	123
27 Wallbreaker	126
28 Technical Artist	131
29 Spider	134
30 Double Surround	139
31 Snaker	142

IMPORTANT NOTES

If you are a newcomer to computing, and are unfamiliar with the task of entering program listings, remember that it is of paramount importance to follow every detail. There are no real short cuts.

Your computer is very particular about grammar and punctuation points and will not run the programs if even a small detail is missing.

Be very fussy about the difference between 0 (zero) and □ (letter O). All the programs in this book have been printed directly from the Amstrad and are therefore 'bug free'.

If you intend to 'save' some of these programs on tape to ensure you don't have to retype them always make sure that your cassette has accepted the information before you move on.

You will find that patience is it's own reward in this high technology world.

ZOMBIE ISLAND



This is going to be a nasty experience for those amongst you who are slow on reaction time.

You are the only survivor of the mad professor's plans to turn humans into zombies and you are being pursued by ten of your old friends who now, unfortunately, have been 'de-humanised'. Your only hope is to lure the zombies into the pits which are spread around the screen. Sounds easy. There is one snag though, or is it several snags, the zombies can move in groups towards you.

How to play

As we said above, try and live as long as possible by moving with the arrow keys or joystick. If you survive you will be offered the choice of continuing to battle with your former pals.

Because of the sprite facilities of the Amstrad you will find that the chances of survival are fairly slim.

The program

```

10 REM ZOMBIE ISLAND
20 GOSUB 520
30 N=10:PIT=10
40 Z=N
50 MODE 1:BORDER 3
60 PAPER 3:PEN 0:LOCATE 1,1:PRINT"
    ZOMBIE ISLAND";
70 X=21:Y=19:LOCATE X,Y:PAPER 0:PEN 1:PR
INT CHR$(249);
80 DIM A(N),B(N):FOR L=1 TO N
90 A(L)=INT(RND(1)*40+1):B(L)=INT(RND(1)
*23+2)
100 XX=A(L):YY=B(L):GOSUB 660:IF DD<>32
THEN GOTO 90
110 LOCATE XX,YY:PEN 2:PRINT CHR$(225);
120 NEXT L
130 FOR L=1 TO PIT
140 XX=INT(RND(1)*40+1):YY=INT(RND(1)*23
+2)
150 GOSUB 660:IF DD<>32 THEN GOTO 140
160 LOCATE XX,YY:PEN 3:PRINT CHR$(231);
170 NEXT L
180 C=0:D=0
190 C=C+1:IF C>N THEN C=1
200 IF A(C)=-1 THEN GOTO 190
210 S=SGN(X-A(C)):T=SGN(Y-B(C))
220 XX=A(C)+S:YY=B(C)+T:GOSUB 660
230 IF DD=231 THEN GOSUB 400:LOCATE A(C)
,B(C):PAPER 0:PRINT " ";A(C)=-1:LOCATE X
,YY:PRINT " ";GOTO 270
240 IF DD=249 THEN GOTO 420
250 LOCATE A(C),B(C):PEN 2:PRINT " ";A(C)
)=XX:B(C)=YY:LOCATE XX,YY:PRINT CHR$(225
);
260 SOUND 0,300,10
270 IF INKEY(0)=0 THEN D=0
280 IF INKEY(2)=0 THEN D=2
290 IF INKEY(1)=0 THEN D=1

```


```

300 IF INKEY(8)=0 THEN D=8
310 S=X:T=Y:IF D=8 AND S>1 THEN S=S-1
320 IF D=1 AND S<40 THEN S=S+1
330 IF D=2 AND T<24 THEN T=T+1
340 IF D=0 AND T>2 THEN T=T-1
350 XX=S:YY=T:GOSUB 660:IF DD=231 THEN G
OTO 420
360 IF DD=225 THEN GOTO 420
370 LOCATE X,Y:PEN 1:PRINT " ";LOCATE S,
T:PRINT CHR$(249);
380 X=S:Y=T
390 GOTO 190
400 Z=Z-1:IF Z=0 THEN GOTO 490
410 FOR L=1 TO 50:SOUND 1,L*10,1:NEXT L:
RETURN
420 FOR L=1 TO 50:SOUND 1,RND(1)*2000+10
0,2:BORDER INT(RND(1)*24)
430 NEXT L
440 LOCATE 1,8:PAPER 3:BORDER 3:PEN 0:PR
INT"          CAUGHT BY THE ROBOTS
"
450 LOCATE 1,25:PAPER 0:PEN 2:PRINT"
PRESS <SPACE> TO PLAY AGAIN      "
460 IF INKEY$=" " THEN GOTO 460
470 IF INKEY$("<")" " THEN GOTO 470
480 RUN
490 REM PLAYER WIN ROUTINE
500 LOCATE 1,8:PAPER 3:BORDER 3:PEN 0:PR
INT"          YOU HAVE SURVIVED THE ISLAND
"
510 GOTO 460
520 REM PUT MACHINE CODE JUST ABOVE RE-A
DJUSTED HIMEM
530 MEMORY FRE(0)-&80
540 MC=HIMEM+1
550 TA=HIMEM+&7F
560 TH=INT(TA/256)
570 TL=TA-256*TH
580 POKE MC,&CD
590 POKE MC+1,&60
600 POKE MC+2,&BB
610 POKE MC+3,&32
620 POKE MC+4,TL
630 POKE MC+5,TH
640 POKE MC+6,&C9
650 RETURN

```

```
660 LOCATE XX,YY  
670 CALL MC  
680 DD=PEEK(TA)  
690 RETURN
```

CLEAR PATH



You may recall that the United States space shuttle Challenger was often nicknamed 'The Flying Brick' by it's pilots as it fell to Earth at 5,000 miles an hour, without an engine.

We have given you the same problem to solve, except we have made it even nastier for you by bringing you into a deserted city.

We have given you one edge however. You have a cargo of bombs which you can use to clear a path for your flying brick.

How to play

Entering the screen at top left you must 'fly' across the city and demolish the skyscrapers by pressing the space bar to release your bombs.

You can only drop a second, or third, bomb when the first has hit it's target so there is no point in pressing the space bar and hoping for the best.

Programming Hints

To change the height of the buildings alter the number after the * sign in line 680. Remember if you increase the height you will reduce the time you have to respond.

The Program

```

10 REM CLEAR PATH
20 GOSUB 440
30 ENT 1,200,2,1
40 BY=-1
50 GOSUB 620
60 LOCATE 1,1:PAPER 5:PEN 10
70 PRINT"          SCORE :          ";
80 X=1:Y=3
90 LOCATE 11,1
100 PAPER 5:PEN 10:PRINT SCORE;
110 PAPER 0
120 TX=X+1:TY=Y
130 IF TX=20 THEN TX=1:TY=Y+1
140 IF TY=25 AND TX=10 THEN LOCATE 5,4:P
EN 13:PRINT"LANDED SAFELY":SC=SC+200:FOR
  W=1 TO 1300:NEXT W:MODE 0:GOTO 50
150 XX=TX+1:YY=TY:IF XX=21 THEN XX=1:YY=
YY+1
160 GOSUB 580
170 IF DD=32 THEN GOTO 250
180 PEN 8:LOCATE 5,4:PRINT"YOU CRASHED"
190 IF INKEY$=" " THEN GOTO 190
200 IF INKEY$(">") THEN GOTO 200
210 RUN
220 SOUND 129,0:SOUND 130,0
230 FOR L=15 TO 0 STEP -1
240 SOUND &129,0,3,5,0,0,L:NEXT L
250 LOCATE X,Y:PRINT" ";
260 LOCATE TX,TY:PRINT CHR$(197);CHR$(20
2);

```

```


270 X=TX:Y=TY
280 IF INKEY(47)<>0 OR BY<>-1 THEN GOTO
330
290 SOUND 1,10,280,4,0,1,0
300 BX=X+1
310 BY=Y+1
320 IF BY=26 THEN BY=-1
330 IF BY=-1 THEN GOTO 430
340 IF BY=25 THEN LOCATE BX,BY:PRINT " ";
:BY=-1:SOUND 130,0:SOUND 129,0:GOTO 430
350 XX=BX:YY=BY+1
360 GOSUB 580
370 IF DD=32 THEN LOCATE BX,BY:PRINT " ";
:LOCATE BX,BY+1:PEN 9:PRINT CHR$(252);:B
Y=BY+1:GOTO 430
380 SOUND 129,0
390 SOUND 130,0,100,3,0,0,RND(1)*4+10
400 SCORE=SCORE+10
410 LOCATE BX,BY:PRINT " ";
420 BY=BY+1:LOCATE BX,BY:PEN 9:PRINT CHR
$(252);
430 GOTO 90
440 REM PUT MACHINE CODE JUST ABOVE RE-A
DJUSTED HIMEM
450 MEMORY FRE(0)-&80
460 MC=HIMEM+1
470 TA=HIMEM+&7F
480 TH=INT(TA/256)
490 TL=TA-256*TH
500 POKE MC,&CD
510 POKE MC+1,&60
520 POKE MC+2,&BB
530 POKE MC+3,&32
540 POKE MC+4,TL
550 POKE MC+5,TH
560 POKE MC+6,&C9
570 RETURN
580 LOCATE XX,YY
590 CALL MC
600 DD=PEEK(TA)
610 RETURN
620 REM DRAW THE CITY-SCAPE
630 MODE 0
640 BORDER 7
650 PEN 3
660 PAPER 1

```



```
670 FOR L=1 TO 20
680 FOR Q=25 TO 24-RND(1)*4 STEP -1
690 LOCATE L,Q
700 PRINT CHR$(207);
710 NEXT Q
720 NEXT L
730 RETURN
```

CASTLE 'ADVENTURE'



Start looking out your armour, your good luck charms and your trusty friends as you are on your way to try and save the Elf King's daughter, yes we said Elf, from her prison.

The girl's father is waiting for her return in the woods outside the castle. He's not stupid!

How to play

The program only uses two word commands, all of which are shown in the data statements at the end of the listing. Do not try to enter multiple statements such as "TAKE KNIFE AND RUN".

You can, however, abbreviate the direction commands to their initial letter i.e. "GO N" for Go North.

If you haven't 'adventured' before remember that, if you are stuck somewhere and can't seem to be able to move on, you

can ask to take Inventory which can also be abbreviated to INV.

The Program

```

1  REM CASTLE ADVENTURE
5  MODE 2
10 DIM DD$(50),OP(18),OD$(18),STUNT(50)
20 DIM ROUTE(50,6)
30 PLOC=1
40 PF=0
90 GOSUB 9000
95 GOSUB 5000
100 GOSUB 7600
110 GOSUB 3000:IF VERB=0 THEN PRINT"I do
n't understand you":GOTO 100
115 GOSUB 4000
120 ON VERB GOSUB 7000,7000,7000,7000,70
00,7000,8000,6000,6500,5000,7500,7400,73
00,7700,7800,7900
130 GOTO 100
999 END
1000 REM RETURN VERB NUMBER IN I/P STRIN
G
1010 RESTORE 10000:T=0
1020 READ T$
1030 IF T$="*" THEN RETURN
1040 IF LEFT$(T$,1)="!" THEN T$=RIGHT$(T
$,LEN(T$)-1):T=T+1
1050 IF T$=V$ THEN VERB=T:RETURN
1060 GOTO 1020
2000 REM RETURN NOUN NUMBER IN I/P STRIN
G
2010 RESTORE 11000:T=0
2020 READ T$
2030 IF T$="*" THEN RETURN
2040 IF LEFT$(T$,1)="!" THEN T$=RIGHT$(T
$,LEN(T$)-1):T=T+1
2050 IF T$=N$ THEN NOUN=T:RETURN
2060 GOTO 2020
3000 REM PARSE THEN INPUT AND ANALYSE
3010 VERB=0:NOUN=0:IF I$="" THEN RETURN

```

```

3020 P=0:FOR L=LEN(I$) TO 1 STEP -1
3030 IF MID$(I$,L,1)=" " THEN P=L
3040 NEXT L
3050 IF P=0 THEN V$=I$:GOSUB 1000:RETURN
3060 V$=MID$(I$,1,P-1):N$=MID$(I$,P+1,40
)
3070 GOSUB 1000:GOSUB 2000
3099 RETURN
4000 ON STUNT(PLOC)+1 GOTO 4010,4100,420
0,4300,4400,4500,4600,4700,4800,4900
4010 RETURN
4100 IF PF<>1 AND OP(3)=0 THEN OP(3)=-1:
PRINT"The princess dies of hunger":GOTO
4010
4110 RETURN
4200 PRINT"You fall out of the tree and
break your neck":GOTO 8990
4300 PRINT"The tree is wobbling very dan
gerously":stunt(ploc)=2:GOTO 4010
4400 PRINT"The tree is swaying in the wi
nd":stunt(ploc)=3:GOTO 4010
4500 PRINT""That looks nice", says the g
uard, eyeing your gold":GOTO 4010
4700 IF op(12)<>0 THEN PRINT"The floor d
rags you back, faster than you can move"
:ploc=46:GOTO 4010
4710 PRINT"Your ring of speed glows and
you rush down the corridor past the slid
ing floor":ploc=50:GOTO 4010
4800 PRINT"A portcullis drops down hard
behind you, barring the castle entrance"
:stunt(ploc)=0:GOTO 4010
4900 IF op(3)<>0 THEN PRINT"The elven ki
ng sees that you have failed in your res
cue attempt, and he swiftly kills you":G
OTO 8990
4910 PRINT"The elven king greets you wit
h fair words and offers you all the gold
and silver of his kingdom as reward for
the safe re
turn of his daughter"
4920 GOTO 8990
5000 REM describe (ploc)
5010 PRINT"You are ";
5020 I=0
5030 I=I+1:IF I>LEN(DD$(ploc)) THEN GOTO
5099

```

```

5040 IF MID$(DD$(PLOC),L,1)="!" OR MID$(
DD$(PLOC),L,1)="$" OR MID$(DD$(PLOC),L,1
)=" " THEN GOTO 5099
5050 PRINT MID$(DD$(PLOC),L,1);
5060 GOTO 5030
5099 PRINT:GOSUB 5200:RETURN
5100 REM DESCRIBE OBJECT NUMBER (N)
5110 PRINT
5120 L=0
5130 L=L+1:IF L>LEN(OD$(N)) THEN GOTO 51
99
5140 IF MID$(OD$(NOUN),L,1)=" " OR MID$(O
D$(NOUN),L,1)="!" THEN GOTO 5199
5150 PRINT MID$(OD$(N),L,1);
5160 GOTO 5130
5199 PRINT:RETURN
5200 REM DESCRIBE ALL (PLOC) OBJECTS
5205 F=0:PRINT"There is ";
5210 FOR N=1 TO 18
5220 IF OP(N)=PLOC THEN NOUN=N:F=1:GOSUB
5100
5230 NEXT N:PRINT:IF F=0 THEN PRINT"noth
ing here" ELSE PRINT "here"
5299 RETURN
6000 REM TAKE COMMAND
6010 IF NOUN=0 THEN PRINT"Take what ":R
ETURN
6015 IF OP(NOUN)<>PLOC OR OP(NOUN)=-1 TH
EN PRINT"Take what ":RETURN
6020 IF OP(NOUN)=0 THEN PRINT"You've alr
eady got it, dumbo":RETURN
6025 IF NOUN>13 THEN PRINT"You can't tak
e that":RETURN
6030 PRINT"D.K. You've got it now"
6040 OP(NOUN)=0:RETURN
6500 REM DROP COMMAND
6510 IF NOUN=0 THEN PRINT"Drop what ":R
ETURN
6515 IF OP(NOUN)<>0 THEN PRINT"Drop what
?":RETURN
6520 PRINT"D.K. It's dropped"
6530 OP(NOUN)=PLOC:RETURN
6700 FOR L=1 TO LEN(OD$(NOUN))
6710 IF MID$(OD$(NOUN),L,1)="!" THEN OD$
(NOUN)=LEFT$(OD$(NOUN),L-1)+" "+MID$(OD$
(NOUN),L+1)

```

```

6720 NEXT L:RETURN
6800 FOR L=1 TO LEN(DD$(PLOC))
6810 IF MID$(DD$(PLOC),L,1)="!" THEN DD$(PLOC)=LEFT$(DD$(PLOC),L-1)+" "+MID$(DD$(PLOC),L+1)
6820 NEXT L:RETURN
6900 FOR L=1 TO LEN(DD$(PLOC))
6910 IF MID$(DD$(PLOC),L,1)="$" THEN DD$(PLOC)=LEFT$(DD$(PLOC),L-1)+" "+MID$(DD$(PLOC),L+1)
6920 NEXT L:RETURN
7000 NLOC=ROUTE(PLOC,VERB):IF NLOC<>-1 AND NLOC<>0 THEN PRINT"O.K.":PRINT:PLOC=NLOC:GOSUB 5000:GOSUB 4000:RETURN
7010 IF NLOC=-1 THEN GOTO 7030
7020 IF VERB=5 THEN PRINT"There's nothing that you can climb here":RETURN
7025 PRINT"You can't move in that direction":RETURN
7030 PRINT"A door blocks your way":RETURN
7300 IF NOUN<>0 THEN PRINT"try EXAMINE":RETURN
7310 GOSUB 6800:GOSUB 5000:RETURN
7400 IF NOUN=0 THEN PRINT"examine what ?":RETURN
7405 IF OP(NOUN)<>0 AND OP(NOUN)<>PLOC THEN PRINT"examine what ?":RETURN
7410 GOSUB 6700:N=NOUN:GOSUB 5100:RETURN
7500 REM INVENTORY
7510 F=0:PRINT"You are holding :":FOR N=1 TO 18
7520 IF OP(N)=0 THEN F=1:NOUN=N:GOSUB 5100
7530 NEXT N:PRINT:IF F=0 THEN PRINT"nothing"
7599 RETURN
7600 I$="":PRINT:PRINT" : ";
7610 A$=INKEY$:IF A$<>" " THEN GOTO 7610
7620 A$=INKEY$:IF A$="" THEN GOTO 7620
7625 IF A$>="a" AND A$<="z" THEN A$=CHR$(ASC(A$)-32)
7630 IF ASC(A$)=13 AND LEN(I$)>0 THEN PRINT:RETURN
7640 IF ASC(A$)<>127 AND LEN(I$)<30 THEN I$=I$+A$:PRINT A$;:GOTO 7610

```

```

7650 IF ASC(A$)<>127 THEN GOTO 7610
7660 IF LEN(I$)=0 THEN GOTO 7600
7670 I$=LEFT$(I$,LEN(I$)-1):PRINT CHR$(8)
);" ";CHR$(8);:GOTO 7610
7700 IF NOUN=0 THEN PRINT"Give what ?":R
ETURN
7710 IF NOUN<>2 AND NOUN<>7 AND NOUN<>9
AND NOUN<>4 AND NOUN<>12 THEN PRINT"It's
  returned to you":RETURN
7720 IF PLOC<>17 THEN GOTO 7750
7730 IF NOUN<>7 AND NOUN<>12 THEN PRINT"
The guard recognises the stolen goods an
d promptly kills you.":GOTO 8990
7740 PRINT"The guard snatches your bribe
  and opens the gate":OP(NOUN)=-1:ROUTE(1
7,2)=18:STUNT(17)=0:RETURN
7750 IF PLOC<11 OR PLOC>14 THEN PRINT"Th
ere is nobody here who wants anything":R
ETURN
7760 IF NOUN=4 THEN PRINT"The princess t
hanks you kindly for the food and looks
much better":OD$(4)="a beautiful, health
y looking pr
incess":PF=1
7770 OP(NOUN)=-1:RETURN
7800 IF NOUN=0 THEN PRINT"but who ?":RET
URN
7810 IF NOUN<>15 AND NOUN<>16 THEN PRINT
"You can't.":RETURN
7820 IF OP(NOUN)<>PLOC THEN PRINT"You ca
n't.":RETURN
7830 IF NOUN=15 THEN PRINT"The kings esc
ort instantly rises from his light slumb
er and despatches you with one blow of h
is sword":GO
TO 8990
7840 IF OP(5)=0 THEN PRINT"The guard swi
ngs his mace and your knife breaks":OP(5
)=-1:RETURN
7850 PRINT"The guard hits you over the h
ead and knocks you out. He takes the pri
ncess back to the tower"
7860 OP(3)=14:OD$(3)="A very distressed
princess":RETURN
7900 IF NOUN<>0 THEN PRINT"O.K.  '";OD$(
NOUN);"'":RETURN

```

```

7910 IF N$="CPC" AND PLOC=6 THEN ROUTE(6
,3)=7:GOSUB 6900:RETURN
7920 IF N$="464" AND PLOC=3 THEN ROUTE(3
,2)=4:GOSUB 6900:RETURN
7930 PRINT"O.K. ?";N$;"?":RETURN
8000 IF NOUN=0 THEN PRINT"Use what ?":RE
TURN
8005 IF OP(NOUN)<>PLOC AND OP(NOUN)<>0 T
HEN PRINT"Use what ?":RETURN
8010 IF NOUN=1 THEN GOTO 8100
8011 IF NOUN=4 THEN GOTO 8200
8012 IF NOUN=8 THEN GOTO 8300
8013 IF NOUN=13 THEN GOTO 8400
8014 IF NOUN=18 THEN GOTO 8800
8019 GOTO 8750
8100 IF PLOC<>15 THEN PRINT"The ground i
s too hard here":RETURN
8110 PRINT"You dig in the soft earth and
  reveal something":ROUTE(15,6)=48:GOSUB
6900:RETURN
8200 PRINT"Someone, somewhere, needed th
at food more than you did":OP(4)=-1:RETU
RN
8300 PRINT"You'll need to use some rope"
:RETURN
8400 IF PLOC=2 THEN PRINT"The door to th
e south is now open":ROUTE(PLOC,2)=16:RE
TURN
8410 IF PLOC=3 THEN PRINT"A combination
is needed to get into the garden shed":R
ETURN
8420 IF PLOC=6 THEN PRINT"You'll need th
e password to get in here":RETURN
8430 IF PLOC=18 THEN PRINT"There's no wa
y you can unlock a portcullis":RETURN
8440 PRINT"There's nothing to unlock her
e":RETURN
8750 PRINT"You can't use that !":RETURN
8800 PRINT"You carefully tie the bucket
to the rope and lower it down the well":
FOR l=1 TO 800:NEXT l
8810 PRINT>Your draw the bucket up again
":FOR l=1 TO 800:NEXT l:PRINT"There is s
omething at the bottom of it":OP(7)=PLOC
8820 PRINT"You let go of the rope and th
e bucket falls back down the well":op(8)

```



```

--1:op(18)--1:PRINT"The object you found
  lies glitte
ring on the ground"
8830 RETURN
8990 PRINT:PRINT:PRINT:PRINT:PRINT:END
9000 REM SETUP ALL THE ARRAYS
9010 FOR L=1 TO 50:STUNT(L)=0:NEXT L
9020 STUNT(7)=1:STUNT(22)=9:STUNT(28)=4:
STUNT(31)=4
9030 STUNT(17)=5:STUNT(49)=7:STUNT(18)=8
9050 RESTORE 12000:FOR L=1 TO 50
9060 FOR N=1 TO 6:READ ROUTE(L,N)
9070 NEXT N:NEXT L
9080 RESTORE 13000
9090 FOR L=1 TO 18:READ OP(L):NEXT L
9100 OD$(1)="A shovel!ready to dig the e
arth":OD$(2)="A pile of treasure!gathere
d in the days of great empires"
9110 od$(3)="A princess!as beautiful as
any elven king's daughter could be":od$(
4)="Some food!lovingly prepared and fit
for royalty"
9120 od$(5)="A knife!razor sharp, cursed
  by many, loved by few":od$(6)="A large
book!telling of the prowess of the kings
  escort"
9130 od$(7)="A gold coin":od$(8)="A buck
et!of untold riches":od$(9)="many Silver
  Plates!inscribed with the mystical rune
s E.P.N.S."
9140 od$(10)="Notepaper!bearing the lege
nd CPC":od$(11)="An ancient scroll!readi
ng 'find my daughter and be richly rewar
ded'"
9150 od$(12)="A shining ring!of speed":o
d$(13)="A key":od$(14)="A tree!branches
hang down low and the trunk reaches for
the sky"
9160 od$(15)="The kings escort!sleeping
lightly":od$(16)="A castle guard!who loo
ks very bored"
9170 od$(17)="A well!radiating magice":o
d$(18)="A long well made rope"
9210 dd$(1)="in the centre of the castle
  courtyard":dd$(2)="at the south end of
the courtyard!an opening leads further s
outh"

```

9220 dd\$(3)="by the entrance to the west tower the garden shed is south\$the shed door is open":dd\$(4)="inside a garden shed"

9230 dd\$(5)="at the bottom of a winding stairway":dd\$(6)="by the entrance to the east tower":dd\$(7)="at the bottom of a staircase"

9240 dd\$(8)="on a winding stairway":dd\$(9)=dd\$(8):dd\$(10)="at the top of a stairway! '464' is scrawled on the wall"

9250 dd\$(11)="on a steep staircase":dd\$(12)=dd\$(11):dd\$(13)=dd\$(11):dd\$(14)="at the very top of the east tower"

9260 dd\$(15)="beside a soft patch of earth in the courtyard\$a hole leads down":dd\$(16)="at the northern end of a guarded passage"

9270 dd\$(17)="in a tunnel. An aging guard blocks your way south":dd\$(18)="on the main East/West road of middle earth"

9280 dd\$(19)="on the road, a forest lies to the south":dd\$(20)="on the road. A large sign reads : Adventure in need of extension :

: ROAD AHEAD CLOSED :

9290 dd\$(21)="on top of a large unclimable cliff overlooking a vast panorama of places to adventure":dd\$(22)="in a small clearing in the forest"

9300 dd\$(23)="on the edge of a vast forest":dd\$(24)="lost deep in a forest":dd\$(25)="lost in a forest":dd\$(26)="lost with him in a forest"

9310 dd\$(27)="halfway up a large tree! the tree extends upwards":dd\$(28)="at the top of the tree, a small clearing is visible below"

9320 dd\$(29)="in the courtyard":dd\$(30)="in the shade in the courtyard":dd\$(31)="at the top of a tree in the courtyard there is a princess in the east tower"

```

9330 dd$(32)="In the north-west corner o
f the courtyard":dd$(33)="on stairs lead
ing down into the dungeon":dd$(34)="in t
he guard roo
m"
9340 dd$(35)="In the banqueting hall":dd
$(36)="At the east end of the banqueting
hall!a door leads north to the kitchen"
9350 dd$(37)="In the castle kitchen":dd$
(38)="At the north end of a north-south
corridor":dd$(39)="in the servants quart
ers"
9360 dd$(40)="in the royal bedroom!there
is a door to the west":dd$(41)="in the
castle library":dd$(42)="in the grand th
rone room"
9370 dd$(43)="in a north-south corridor!
cells lead east and west":dd$(44)="in an
empty cell":dd$(45)=dd$(44)
9380 dd$(46)="in a north-south corridor"
:DD$(47)="in a cell"
9390 dd$(48)="in a locked dungeon cell":
dd$(49)="At the south end of a corridor"
:dd$(50)=dd$(49)
9999 RETURN
10000 DATA !NORTH,N,!SOUTH,S,!EAST,E,!WE
ST,W
10010 DATA !UP,U,CLIMB,!DOWN,D
10020 DATA !USE
10030 DATA !TAKE,GET
10040 DATA !DROP,PUT
10050 DATA !LOOK
10060 DATA !INVENTORY,INV
10070 DATA !EXAMINE,!SEARCH
10080 DATA !GIVE,BRIBE,!FIGHT,KILL,ATTAC
K
10090 DATA !SAY,SHOUT
10999 DATA*
11000 DATA !SHOVEL,SPADE,!TREASURE,!PRIN
CESS,!FOOD
11010 DATA !KNIFE,WEAPON,!BOOK,!GOLD,COI
N,!BUCKET,!SILVER,PLATES
11020 DATA !NOTEPAPER,PAPER,NOTE,!SCROLL
,!RING,SPEED,!KEY
11030 DATA !TREE,TRUNK,!ESCORT,!GUARD,SO
LDIER,BORED,!WELL,!ROPE


```

```

11999 DATA*
12000 DATA 29,2,6,3,0,0,1,-1,30,0,0,0,32
,-1,1,5,0,0
12010 DATA 3,0,0,0,0,0,0,0,3,0,8,0,15,30
,-1,1,0,0
12020 DATA 0,0,0,6,11,0,0,0,0,0,9,5,0,0,
0,0,10,8,0,0,0,0,9,0,0,0,0,12,7,0,0,0,
0,13,11
12030 DATA 0,0,0,0,14,12,0,0,0,0,0,13,35
,6,0,29,0,0,2,17,0,0,0,0,16,0,0,0,0,0,-1
,23,19,21,0,0
12040 DATA 0,23,20,18,0,0,0,23,0,19,0,0,
0,23,18,0,0,0
12050 DATA 25,26,24,26,0,0,18,24,26,25,0
,0,25,26,24,25,27,0,26,24,25,26,27,0,25,
24,26,25,27,0,0,0,0,0,28,22
12060 DATA 0,0,0,0,0,27,0,1,15,32,0,0,6,
0,0,2,31,0,0,0,0,0,30,34,3,29,0,0,0,0,
0,35,40,0,38
12070 DATA 42,32,0,0,0,0,39,15,36,33,0,0
,37,0,0,35,0,0,0,36,0,39,0,0,0,43,-1,-1,
33,0,0,35,37,0,0,0
12080 DATA 41,0,-1,-1,0,0,0,40,0,42,0,0,
0,34,41,0,0,0,38,46,44,45,0,0,0,0,0,43,0
,0,0,0,43,0,0,0
12090 DATA 43,49,47,-1,0,0,0,0,0,46,0,0,
0,0,-1,0,15,0,0,0,0,0,0,0,46,0,0,0,0,0
13000 DATA 4,42,14,37,39,41,-1,29,35,50,
0,48,47,30,34,17,29,29

```

MINEFIELD



It's a darned hard life for the little man who has been deserted by his fellow soldiers, in the middle of a mine field. See if you can help him reach the dug-out where his colleagues are hiding.

The mines are randomly generated, so be careful of your moves. Don't tell the little soldier but we believe, on very good authority, that when he reaches the safety of the dug-out he is going to be sent out on another mission.

Every time he makes it through the minefield the number of mines increase and he must keep going until

How to play

Use arrow keys to move. When the inevitable overcomes your hero you will see a score on the screen and you may press the space bar to send the poor little beggar back into the field.

Programming hints

Set the difficulty level when asked to do so.

The program

```

10 REM MINEFIELD
20 GOSUB 550
30 C=5
40 D=1
50 GOSUB 430
60 SCORE=0
70 MODE 0
80 BORDER 13
90 PAPER 13:PEN 5
100 LOCATE 1,1
110 PRINT"      MINEFIELD          SCORE
    :          ";
120 LOCATE 1,25:PRINT"
    ";
130 PAPER 0
140 EVERY 4,1 GOSUB 930
150 PEN 14
160 LOCATE 19,23
170 PRINT CHR$(203);
180 X=3:Y=3
190 XX=X:YY=Y
200 Q=DIF
210 Q=Q-1
220 IF Q<>0 THEN GOTO 210
230 SCORE=SCORE+10
240 IF INKEY(8)=0 AND XX<>1 THEN XX=XX-1
250 IF INKEY(1)=0 AND XX<>20 THEN XX=XX+
1
260 IF INKEY(0)=0 AND YY<>3 THEN YY=YY-1
270 IF INKEY(2)=0 AND YY<>24 THEN YY=YY+
1
280 IF XX=X AND YY=Y THEN GOTO 360
290 GOSUB 690
300 IF DD=231 THEN GOTO 830
310 IF DD=203 THEN GOTO 740
320 IF DD<>32 THEN GOTO 190
330 LOCATE X,Y:PRINT" "
```


```

340 LOCATE XX,YY:PEN 9
350 PRINT CHR$(249);
360 X=XX:Y=YY
370 XX=INT(RND(1)*20+1):YY=INT(RND(1)*22
+3)
380 GOSUB 690
390 IF DD<>32 THEN GOTO 370
400 PEN 8:LOCATE XX,YY
410 PRINT CHR$(231);
420 GOTO 190
430 REM SET DIFFICULTY LEVEL
440 MODE 0
450 LOCATE 1,4
460 PRINT"  SET THE DIFFICULTY"
470 LOCATE 1,8
480 PRINT"0 IS EASY"
490 LOCATE 1,10
500 PRINT"      9 IS HARDEST"
510 K$=INKEY$
520 IF K$<"0" OR K$>"9" THEN GOTO 510
530 DIF=50-5*(ASC(K$)-ASC("0"))
540 RETURN
550 REM PUT MACHINE CODE JUST ABOVE RE-A
DJUSTED HIMEM
560 MEMORY FRE(0)-&80
570 MC=HIMEM+1
580 TA=HIMEM+&7F
590 TH=INT(TA/256)
600 TL=TA-256*TH
610 POKE MC,&CD
620 POKE MC+1,&60
630 POKE MC+2,&BB
640 POKE MC+3,&32
650 POKE MC+4,TL
660 POKE MC+5,TH
670 POKE MC+6,&C9
680 RETURN
690 REM READ FROM SCREEN (XX,YY) INTO DD
700 LOCATE XX,YY
710 CALL MC
720 DD=PEEK(TA)
730 RETURN
740 REM REACHED EDGE OF MINEFIELD
750 LOCATE 1,8
760 PRINT"  YOU HAVE ESCAPED  "
770 LOCATE 1,10

```

```
780 PRINT"  YOU SCORE :
790 LOCATE 15,10
800 PRINT SCORE
810 GOTO 890
820 END
830 REM HIT A MINE => KILLED
840 PAPER 4:PEN 15
850 LOCATE 1,8
860 PRINT"YOU HAVE BEEN KILLED"
870 LOCATE 1,10
880 PRINT"  YOU SCORE 0      "
890 IF INKEY$=" " THEN GOTO 890
900 IF INKEY$("<>") THEN GOTO 900
910 GOTO 60
920 END
930 C=C+D:IF C=25 OR C=4 THEN D=-D:GOTO
930
940 INK 8,C
950 RETURN
```


ASTEROID STORM



Your force field defences have disintegrated; your lasers have overheated and you are now approaching a meteoroid belt.

We don't hold out much hope for your survival as it's only a matter of time, and manual dexterity, as to how long you can keep going.

Your cruiser is moving down the screen and a continuous wave of rocks and star particles are rising to meet you so good luck.

How to play

As you have no defence systems left you must try to stay alive by moving left and right using arrow keys.

Programming Hints

To slow up the asteroids you can insert a delay as follows:

Line 195 FOR Z= 1 to 30: NEXT Z

The Program


```

10 REM ASTEROID STORM
20 GOSUB 410
30 REM NEW PLAYER
40 SCORE=0
50 MODE 1
60 BORDER 18
70 SHIPX=18
80 YY=1
90 SHIP$=" "+CHR$(245)+" "
100 PAPER 0
110 INK 0,13
120 CLS
130 REM MAIN LOOP
140 A=INT(RND(1)*35+1)
150 B=INT(RND(1)*(38-A)+1)
160 PEN 3
170 LOCATE A,25:PRINT"*";
180 LOCATE B,25:PRINT"*":LOCATE 40,25:PR
INT" "
190 REM MOVE SHIP
200 SCORE=SCORE+15
210 IF INKEY(8)=0 AND SHIPX>1 THEN SHIPX
=SHIPX-1
220 IF INKEY(1)=0 AND SHIPX<38 THEN SHIP
X=SHIPX+1
230 XX=SHIPX+1
240 GOSUB 550
250 IF DD=42 THEN GOTO 300
260 LOCATE SHIPX,1
270 PEN 1
280 PRINT SHIP$
290 GOTO 130
300 REM KILLED BY A ASTEREORIC COLLISION
310 PAPER 1
320 PEN 3

```

```
330 LOCATE 16,6:PRINT"Oh Dear,"
340 FOR Q=1 TO 1000:NEXT Q
350 LOCATE 8,10:PRINT"Another one bites
the dust"
360 FOR q=1 TO 1000:NEXT q
370 LOCATE 10,14:PRINT"YOU SCORED : ";SC
ORE;" "
380 IF INKEY$=" " THEN GOTO 380
390 IF INKEY$("<>") THEN GOTO 390
400 GOTO 30
410 REM PUT MACHINE CODE JUST ABOVE RE-A
DJUSTED HIMEM
420 MEMORY FRE(0)-&80
430 MC=HIMEM+1
440 TA=HIMEM+&7F
450 TH=INT(TA/256)
460 TL=TA-256*TH
470 POKE MC,&CD
480 POKE MC+1,&60
490 POKE MC+2,&BB
500 POKE MC+3,&32
510 POKE MC+4,TL
520 POKE MC+5,TH
530 POKE MC+6,&C9
540 RETURN
550 LOCATE XX,YY
560 CALL MC
570 DD=PEEK(TA)
580 RETURN
```

WORDGRAMS



If you are tired of zapping, crunching and powing the aliens of the Universe perhaps you would like to try something a little more cerebral.

Your Amstrad will choose a word at random, from a selection, and scramble the letters on the screen. You must determine the word and enter your answer.

How to play

Type in the word you think has been scrambled and press ENTER.

You will be given a selection of ten words and will receive points based on your efforts.

Programing Hints

You can alter the words by changing the data in lines 290 onwards but remember, if you change the number of words, you must change the items in Line 280.

The Program


```

10 REM ANAGRAMS
20 MODE 2
30 RESTORE
40 READ N
50 FOR L=1 TO INT(RND(1)*N+1)
60 READ W$
70 NEXT L
80 A$=W$
90 FOR L=1 TO 30
100 S=INT(RND(1)*LEN(W$)+1)
110 A$=MID$(A$,S,1)+LEFT$(A$,S-1)+MID$(A$,S+1,255)
120 NEXT L
130 CLS:LOCATE 1,10:PRINT"The anagram is
: ";
140 PRINT a$
150 LOCATE 1,16
160 PRINT"What is the hidden word :
"
170 LOCATE 26,16
180 INPUT g$
190 IF G$="" OR LEN(G$)>15 THEN GOTO 150
200 IF G$=W$ THEN GOTO 250
210 LOCATE 1,20
220 PRINT"WRONG ... BAD LUCK"
230 FOR Q=1 TO 1500:NEXT Q
240 GOTO 30
250 LOCATE 1,20
260 PRINT"CORRECT ... WELL DONE"
270 GOTO 230
280 DATA 40
290 DATA AMSTRAD,COMPUTER
300 DATA TELEVISION,COLOUR
310 DATA CINEMA,BUNGALOW
320 DATA SUITCASE,CASSETTE

```

330 DATA CARPET, CEILING
340 DATA MATHEMATICAL
350 DATA GYROSCOPE
360 DATA AERIAL, GENERATOR
370 DATA GLASS, NEEDLE
380 DATA SIENNA, VIENNA
390 DATA CHARISMATIC, REMOTE
400 DATA PICTURE, ENCYCLOPEDIA
410 DATA RABBIT, AARDVARK
420 DATA STANDARD, TELEPHONE
430 DATA BOXES, BOARD
440 DATA HEATER, AMPLIFIER
450 DATA BASIC, VIDEO
460 DATA DIFFICULT, STRANGE
470 DATA CARAVAN, ALARM
480 DATA SPADE, SKIER
490 DATA HOARSE, FROST

DRIVING TEST



You have been hurtling down country lanes in your new sports car, trying to impress your lady friend by your side, only to find that your brakes have gone and there is a steep incline before you.

Try to stay on the road — you know it makes sense.

How to play

Steer the car using the arrow keys to move left and right. When you finally run out of road you will see a score on the screen.

The program

```


10 REM DRIVING TEST
20 DEFINT A-Z
30 MODE 1
40 SYMBOL 255,&99,&FF,&99,&3C,&3C,&BD,&F
F,&99
50 GOSUB 440
60 LOCATE 5,5:INPUT"DIFFICULTY (4-7) ";D
70 IF D<4 OR D>7 OR D<>INT(D) THEN GOTO
60
80 ROAD$=CHR$(217)+CHR$(207)
90 FOR L=1 TO D
100 ROAD$=ROAD$+" "
110 NEXT L
120 ROAD$=ROAD$+CHR$(207)+CHR$(219)
130 RMAX=37-D
140 CLS
150 SC=0
160 BORDER 21
170 CAR=8:ROAD=12
180 LOCATE 1,25:FOR L=1 TO ROAD:PRINT CH
R$(207);:NEXT L
190 LOCATE ROAD+D+1,25:FOR L=1 TO 28-D:P
RINT CHR$(207);:NEXT L
200 PRINT
210 XX=CAR:YY=2:GOSUB 580
220 SC=SC+1
230 IF DD<>32 THEN GOTO 320
240 IF INKEY(8)=0 AND CAR>1 THEN CAR=CAR
-1
250 IF INKEY(1)=0 AND CAR<40 THEN CAR=CA
R+1
260 LOCATE XX,1:PRINT CHR$(255);
270 IF RND(1)>0.4 AND ROAD>1 THEN ROAD=R
OAD-1
280 IF RND(1)>0.4 AND ROAD<RMAX THEN ROA
D=ROAD+1
290 LOCATE ROAD,25:PRINT ROAD$:PRINT
300 SOUND 1,1000,3,7,0,0,5
310 GOTO 210
320 FOR L=15 TO 0 STEP -1
330 BORDER L*1.5
340 SOUND 1,0,30,7,0,0,L

```



```
350 FOR Q=1 TO 5
360 PLOT 310,200
370 DRAW RND(1)*639+1,RND(1)*400+1
380 NEXT Q
390 NEXT L
400 LOCATE 1,25:PRINT"                YOU S
CORED : ";SC
410 IF INKEY$=" " THEN GOTO 410
420 IF INKEY$("<>") THEN GOTO 420
430 GOTO 60
440 REM PUT MACHINE CODE JUST ABOVE READ
JUSTED HIMEM
450 MEMORY FRE(0)-&80
460 MC=HIMEM+1
470 TA=HIMEM+&7F
480 TH=INT(TA/256)
490 TL=TA-256*TH
500 POKE MC,&CD
510 POKE MC+1,&60
520 POKE MC+2,&BB
530 POKE MC+3,&32
540 POKE MC+4,TL
550 POKE MC+5,TH
560 POKE MC+6,&C9
570 RETURN
580 LOCATE XX,YY
590 CALL MC
600 DD=PEEK(TA)
610 RETURN
```

SPACE TREK



It's time to bravely go where no man has gone before, or something like that anyway.

We have supplied you here with Photon Torpedoes, Phaser Weapons and long and short range scanners but, of course, you are still going to have to deal with the Klingons yourself.

How to play

You will be given a 'menu' to choose from when you begin. Once you have made your choice the playing instructions will appear on the screen.

The Program

```

10 REM SPACE-TREK
20 GOSUB 3000
30 GOSUB 3500
40 EN=5000
50 SD=0
60 NP=10
100 REM MAIN TURN LOOP
105 IF EN<0 THEN GOTO 10000
107 SD=SD+1.04
110 MODE 1:PAPER 0: BORDER 2:CLS
115 PEN 1
120 LOCATE 8,4:PRINT"SPACE TREK COMMANDS
"
130 LOCATE 7,5:PRINT"=====
=="
140 LOCATE 1,11
150 PRINT"(1)  Warp Engines"
160 PRINT"(2)  Impulse Engines"
170 PRINT"(3)  Phasers"
180 PRINT"(4)  Photon Torpedoes"
190 PRINT"(5)  Dock with Starship"
200 PRINT"(6)  Long range scan"
210 PRINT"(7)  Short range scan"
211 A$=INKEY$
213 EN=EN+1:IF EN>5000 THEN EN=5000
215 IF KS=0 THEN LOCATE 5,21:PRINT"CONDI
TION GREEN":GOTO 218 ELSE LOCATE 5,21:PR
INT"CONDITION RED":BORDER 3
216 IF RND<0.06*KS THEN GOSUB 9000
218 LOCATE 15,19:PRINT"SHIP ENERGY : "EN"
"
219 IF A$<"1" OR A$>"7" THEN GOTO 211
220 ON ASC(A$)-ASC("0") GOSUB 4000,5000,
6000,7000,8000,1000,2000
299 GOTO 100
1000 REM LONG RANGE SCAN
1010 CLS
1020 A=SX-1:C=SX+1
1030 B=SY-1:D=SY+1
1040 IF A<1 THEN A=1 ELSE IF C>10 THEN C
=10
1050 IF B<1 THEN B=1 ELSE IF D>10 THEN D
=10

```

```

1055 LOCATE 6,5
1060 FOR L=A TO C
1070 PRINT"***"CHR$(L+47)"***";
1080 NEXT L
1090 LOCATE 1,6:FOR L=B TO D
1100 PRINT"      ";;FOR N=A TO C:PRINT"*
      *";:NEXT N:PRINT
1105 PRINT"      "CHR$(L+47)"      ";;FOR N=A TO
      C:PRINT"*      *";:NEXT N:PRINT
1108 PRINT"      ";;FOR N=A TO C:PRINT"*
      *";:NEXT N:PRINT
1110 NEXT L
1120 PRINT"      ";;FOR L=A TO C:PRINT"***
*****";:NEXT L
1125 LOCATE 4,1:PRINT"<stars , bases , k
lings>"
1130 FOR L=A TO C
1140 FOR M=B TO D
1145 SOUND 1,INT(RND*500)+50,10
1150 LOCATE 8+(L-A)*7,7+(M-B)*3
1160 G$=STR$(G(L,M)):G$=RIGHT$(G$,LEN(G$
)-1):G$=RIGHT$("00"+G$,3)
1165 PRINT G$
1170 NEXT M:NEXT L
1180 SOUND 281,0
1200 LOCATE 6,21:PRINT"<SPACE> to exit"
1210 IF INKEY$="" THEN GOTO 1210
1220 IF INKEY$("<") THEN GOTO 1220
1299 RETURN
2000 REM SHORT RANGE SCAN
2010 CLS:LOCATE 6,6:PRINT"**0**1**2**3**
4**5**6**7**8**9**"
2020 FOR L=0 TO 9:LOCATE 3,L+7
2030 PRINT STR$(L)" *
      *"
2040 NEXT L
2050 LOCATE 6,17
2060 PRINT"*****"
***"
2065 PAPER 1:PEN 0
2070 FOR L=1 TO 10:FOR M=1 TO 10
2075 SOUND 1,L*M*4+200,10
2080 LOCATE 4+L*3,6+M
2090 IF S(L,M)=100 THEN PRINT" * ":GOTO
2140
2100 IF S(L,M)=-1 THEN PRINT"<!>":GOTO 2
140

```

```

2110 IF S(L,M)=10 THEN PRINT">-<":GOTO 2
140
2120 IF S(L,M)=1 THEN PRINT"<=>"
2130 IF S(L,M)=0 THEN PRINT"  "
2140 NEXT M:NEXT L
2142 PAPER 0:PEN 1
2145 SOUND %81,0
2150 LOCATE 6,21:PRINT"<SPACE> to exit"
2160 IF INKEY$=" " THEN GOTO 2160
2170 IF INKEY$<>" " THEN GOTO 2170
2199 RETURN
3000 REM SETUP THE GALAXY
3010 DIM G(10,10),S(10,10)
3020 NK=30:DIM KX(NK),KY(NK)
3030 SX=5:SY=5
3040 X=5:Y=5
3060 FOR N=1 TO NK
3070 L=INT(RND(1)*10+1):M=INT(RND(1)*10+
1)
3080 G(L,M)=G(L,M)+1
3090 NEXT N
3100 FOR L=1 TO 10:FOR M=1 TO 10
3110 G(L,M)=G(L,M)+100*INT(RND*5)
3120 NEXT M:NEXT L
3130 FOR N=1 TO 3
3140 L=INT(RND(1)*10+1):M=INT(RND(1)*10+
1)
3150 G(L,M)=G(L,M)+10
3160 NEXT N
3199 RETURN
3500 REM GENERATE THIS SECTOR
3510 P=G(SX,SY):FOR L=1 TO 10
3520 FOR M=1 TO 10:S(L,M)=0:NEXT M:NEXT
L
3530 S(X,Y)=-1
3540 IF P<100 THEN GOTO 3550
3543 P=P-100
3546 L=INT(RND(1)*10+1):M=INT(RND(1)*10+
1):IF S(L,M)<>0 THEN GOTO 3546
3549 S(L,M)=100:GOTO 3540
3550 IF P<10 THEN KS=P:GOTO 3560
3553 P=P-10
3556 L=INT(RND(1)*10+1):M=INT(RND(1)*10+
1):IF S(L,M)<>0 THEN GOTO 3556
3559 S(L,M)=10:GOTO 3550
3560 IF P=0 THEN RETURN

```

```

3570 P=P-1
3580 L=INT(RND(1)*10+1):M=INT(RND(1)*10+
1):IF S(L,M)<>0 THEN GOTO 3580
3590 S(L,M)=1:GOTO 3560
3599 RETURN
4000 REM WARP ENGINES
4010 IF EN<1000 THEN CLS:LOCATE 1,11:PRI
NT"Warp engines out of commission":FOR L
=1 TO 800:NEXT L:RETURN
4020 CLS:LOCATE 1,9:PRINT"SET TARGET SEC
TOR"
4030 LOCATE 6,13:PRINT" X :   , Y :   "
4040 L=SX-1:M=SY-1
4050 LOCATE 11,13:PRINT L:LOCATE 18,13:P
RINT M
4060 A$=INKEY$:IF A$="" THEN GOTO 4060
4065 IF A$=CHR$(242) THEN L=L-1
4070 IF A$=CHR$(243) THEN L=L+1
4080 IF A$=CHR$(240) THEN M=M-1
4090 IF A$=CHR$(241) THEN M=M+1
4100 IF A$=CHR$(224) THEN GOTO 4200
4110 L=(L+10) MOD 10:M=(M+10) MOD 10
4115 PRINT:PRINT"Energy used :":DX=L-(S
X-1):DY=M-(SY-1):P=20*INT(DX*DX+DY*DY):P
RINT P;"   "
4120 GOTO 4050
4200 IF EN<P THEN PRINT"INSUFFICIENT ENE
RGY REMAINS":FOR L=1 TO 800:NEXT L:RETUR
N
4220 FOR N=50 TO 200 STEP 5:SOUND 1,N,25
:SOUND 2,N+2,15:SOUND 3,N+4,10:NEXT N
4230 PRINT"YOU ARE NOW IN SECTOR ( "L" ,
"  M" )"
4240 SX=L+1:SY=M+1:FOR L=1 TO 800:NEXT L
4250 EN=EN-P:GOSUB 3500:RETURN
5000 REM IMPULSE ENGINES
5010 IF EN<500 THEN CLS:LOCATE 1,11:PRIN
T"Impulse engines out of commission":FOR
L=1 TO 800:NEXT L:RETURN
5020 CLS:LOCATE 1,9:PRINT"SET TARGET COO
RDINATES"
5030 LOCATE 6,13:PRINT" X :   , Y :   "
5040 L=X-1:M=Y-1
5050 LOCATE 11,13:PRINT L:LOCATE 18,13:P
RINT M
5060 A$=INKEY$:IF A$="" THEN GOTO 5060

```

```

5065 IF A$=CHR$(242) THEN L=L-1
5070 IF A$=CHR$(243) THEN L=L+1
5080 IF A$=CHR$(240) THEN M=M-1
5090 IF A$=CHR$(241) THEN M=M+1
5100 IF A$=CHR$(224) THEN GOTO 5200
5110 L=(L+10) MOD 10:M=(M+10) MOD 10
5115 PRINT:PRINT"ENERGY USED:";:DX=L-(X-
1):DY=M-(Y-1):P=3*INT(DX*DX+DY*DY):PRINT
P;"      "
5120 GOTO 5050
5200 IF EN<P THEN PRINT"INSUFFICIENT ENE
RGY IS LEFT":FOR L=1 TO 800:NEXT L:RETUR
N
5220 PRINT"YOU HAVE MOVED : "
5230 S(X,Y)=0:X=L+1:Y=M+1
5240 IF S(X,Y)<>0 THEN PRINT:PRINT"...I
N FLIGHT COLLISION..." ELSE GOTO 5500
5250 IF S(X,Y)=1 THEN PRINT"..WITH A KLI
NGON":EN=EN-1500:GOTO 5500
5260 IF S(X,Y)=10 THEN PRINT"..WITH A ST
ARBASE":GOTO 8500
5270 IF S(X,Y)=100 THEN PRINT"..WITH A S
TAR..":EN=0:GOTO 5500
5500 S(X,Y)=-1:EN=EN-P:FOR L=1 TO 600:NE
XT L:RETURN
6000 REM PHASERS
6005 DEG
6010 CLS:LOCATE 1,9:PRINT"PHASER WEAPONS
"
6020 LOCATE 1,11:PRINT"BEARING :
      ":LOCATE 11,11:INPUT BR
6030 IF BR<0 OR BR>360 THEN SOUND &81,50
0,50:FOR L=1 TO 800:NEXT L:GOTO 6020
6040 LOCATE 1,14:PRINT"ENERGY :
      ":LOCATE 11,14:INPUT EP
6050 IF EP<0 OR EP>EN THEN SOUND &81,400
,50:FOR L=1 TO 800:NEXT L:GOTO 6040
6060 TX=X:TY=Y:CX=SIN(BR):CY=COS(BR)
6080 TX=TX+CX:TY=TY+CY
6085 IF TX<1 OR TX>10 OR TY<1 OR TY>10 T
HEN GOTO 6200
6090 P=S(INT(TX+0.5),INT(TY+0.5)):IF P=0
THEN GOTO 6080
6100 IF P=10 THEN PRINT"YOU JUST DESTROY
ED A STARBASE":G(SX,SY)=G(SX,SY)-10:S(IN
T(TX),INT(TY))=0:GOTO 6080

```

```

6110 IF P<>1 THEN GOTO 6120
6112 PRINT"YOU HIT A KLINGON"
6115 IF RND(1)*500<EP THEN PRINT"AND YOU
  DESTROYED IT":G(SX,SY)=G(SX,SY)-1:S(INT
  (TX+0.5),INT(TY+0.5))=0:NK=NK-1:KS=KS-1
6120 GOTO 6080
6200 EN=EN-EP:FOR L=1 TO 1000:NEXT L:RET
URN
7000 REM PHOTON TORPEDOES
7010 CLS:LOCATE 1,9:PRINT"PHOTON TORPEDO
ES"
7020 LOCATE 1,11:PRINT"NUMBER TO FIRE :
      ":LOCATE 17,11:INPUT PP
7030 IF PP<1 OR PP>NP THEN SOUND &81,400
,50:FOR L=1 TO 800:NEXT L:GOTO 7020
7040 NP=NP-PP
7050 FOR M=1 TO PP
7060 PRINT"PHOTON TORPEDO "M" FIRED AND
RUNNING.....":FOR L=1 TO 10:PRINT". ";
:SOUND 1,15*L+100:NEXT L:FOR L=1 TO 600:
NEXT L
7065 PRINT
7070 IF RND(1)<0.3 THEN PRINT"IT MISSED.
.....":GOTO 7200
7080 IF G(SX,SY)/10=INT(G(SX,SY)/10) THE
N PRINT"THE TORPEDO RUNS OUT INTO SPACE"
:GOTO 7200
7090 PRINT"HOMING DEVICES ACTUATED"
7100 A=INT(RND(1)*10+1):B=INT(RND(1)*10+
1):IF S(A,B)<>1 THEN GOTO 7100
7110 PRINT"YOU HIT A KLINGON":IF RND(1)<
0.75 THEN PRINT"AND YOU DESTROYED IT":S(
A,B)=0:KS=KS-1:G(SX,SY)=G(SX,SY)-1:GOTO
7200
7120 PRINT"BUT THE KLINGON'S DEFLECTORS
REDUCED THE DAMAGE"
7200 NEXT M
7210 FOR L=1 TO 800:NEXT L
7299 RETURN
8000 REM DOCK WITH STARBASE
8010 CLS:LOCATE 1,6:PRINT"DOCKING COMPUT
ER D.S. 0.1"
8020 PRINT"=====
8030 LOCATE 1,9:PRINT"...A LITTLE UNRELI
ABLE"
8040 LOCATE 1,11:FOR L=1 TO 20+INT(RND(1
)*20)

```



```

8050 A=INT(RND(1)*10+1):B=INT(RND(1)*10+
1)
8060 IF S(A,B)=10 THEN PRINT"DOCKED SUCC
ESSFULLY":EN=5000:NP=10:PRINT:PRINT"ENER
GY AND PHOTON TORPS. RENEWED"
8070 NEXT L
8080 FOR L=1 TO 800 STEP 4:SOUND 1,L,2
8090 NEXT L
8099 RETURN
9000 REM A KLINGON ATTACKS
9010 LOCATE 1,21:PRINT"A KLINGON PHASER
IS FIRED AT YOU"
9013 FOR L=15 TO 0 STEP -1
9015 SOUND 1,0,10,L,0,0,8:NEXT L
9020 IF RND(1)<0.4 THEN PRINT"IT MISSES"
:FOR L=1 TO 600:NEXT L:LOCATE 1,21:PRINT
"
":RETUR
N
9030 PRINT"IT HITS YOU...":DM=INT(RND(1)
*300+400):FOR L=1 TO 600:NEXT L
9040 EN=EN-DM
9050 LOCATE 1,21:PRINT"
":RETURN
10000 REM SHIP HAS BEEN DESTROYED
10010 CLS:LOCATE 1,9:PRINT"YOUR SPACE SH
IP HAS RUN OUT OF ENERGY"
10020 PRINT:PRINT"YOUR LIFE SUPPORT SYST
EMS HAVE FAILED"
10030 PRINT:PRINT".... ALL PASSENGERS AN
D CREW"
10040 PRINT:PRINT"...ARE MISSING ... PRE
SUMED DEAD"
10050 FOR L=1 TO 600:NEXT L
10060 PRINT"THE STARDATE IS : ";SD
10070 LOCATE 1,21:PRINT" AGAIN (Y/N) ?
"
10080 A$=INKEY$:IF A$<>"" THEN GOTO 1008
0
10090 A$=INKEY$:IF A$="" THEN GOTO 10090
10100 IF A$="Y" OR A$="y" THEN RUN
10110 IF A$<>"N" AND A$<>"n" THEN GOTO 1
0080
10120 CLS:LOCATE 1,5:END

```

DRAGON DUNGEON



We have been saving this monstrous program until the middle of the book as we didn't want to scare you with it's length or it's content.

You may have heard of the highly popular Dragons and Dungeons game but have never played it – until now.

You will begin by finding yourself on the third level of a dungeon and, should you proceed into the heart of the maze, you will meet **twenty one** different types of monsters. Lurking in the lower depths are trolls, orcs, snakes, dwarfs, wolves, newts, many others and, most horrible of all, a Dragon capable of inflicting fatal wounds.

On your side you have help, if you can find your way, from a fire demon and from several types of spells. Be careful how you use the spells, however, as they can sometimes break down and leave you deeper in the

How to play

Press the first letter of the command of your choice as the new menu is offered to you.

You will be told of the consequences of your choice and then you are on your own.

By the way, if you live, a higher score will be given.

The Program

```

10 REM DRAGON DUNGEON
15 MODE 2
20 GOSUB 2900:GOSUB 1600:I$=""
25 GOTO 100
30 FOR ZQ=1 TO 500:NEXT ZQ:RETURN
100 REM MAIN NEW LOCATION LOOP
110 GOSUB 2500:GOSUB 30
115 IF LV=0 THEN GOSUB 1500:GOTO 900
120 IF MF=1 THEN GOTO 200
130 GOSUB 1000
140 GOSUB 3400:IF I$="L" THEN PRINT"That
    sounds interesting":GOSUB 30:GOTO 130
142 IF I$="P" THEN PRINT"ho hum....":GOS
UB 30:PRINT"la di da...":GOSUB 30:MF=1:P
M=PM+INT(RND(1)*4):PS=PS+4:GOTO 200
144 IF I$="T" THEN MM=1:PRINT"This can b
e a little unpredictable":GOSUB 30:GOSUB
2800:GOTO 100
150 F=0:FOR L=1 TO LEN(MO$):IF MID$(MO$,
L,1)=I$ THEN F=1
160 NEXT L:IF F=0 THEN PRINT"You can't m
ove in that direction":GOSUB 30:GOTO 130
170 IF I$="U" THEN LV=LV-1:PRINT"You'll
get out of here yet":GOTO 195
180 IF I$="D" THEN LV=LV+1:PRINT"Off goe
s the hero, in search of yet more advent
ure":GOTO 195
190 PRINT"The road goes ever on, anon."
195 GOSUB 30:GOTO 100
200 REM Monster encountered here

```

```

210 PRINT"You just bumped into something
":GOSUB 30
220 IF RND(1)<0.02 THEN df=2:GOSUB 3000:
PRINT"You're not going to believe this,
it's the DRAGON":GOSUB 30:PRINT"Have fun
.....":GOTO
240
230 GOSUB 1700
240 GOSUB 30
250 GOSUB 1100:GOSUB 3400:IF i$<>"S" AND
I$<>"D" AND I$<>"R" AND I$<>"M" THEN GO
TO 250
260 IF I$="S" THEN WP=1:GOSUB 2000
270 IF I$="D" THEN WP=2:GOSUB 2000
280 IF I$="R" AND RND(1)<0.35 THEN PRINT
"You got away from ":PRINT"the ";DJ$:GOT
0 450
290 IF i$="R" THEN PRINT"You couldn't ge
t away from it"
300 IF i$<>"M" THEN GOTO 320
310 GOSUB 2200:ON op GOSUB 2800,2700,210
0
320 IF mf<>0 THEN GOTO 490
330 IF df=2 THEN df=1:GOSUB 1200
340 GOTO 100
450 REM Player left DRAGON wounded
460 IF df=2 THEN df=0:ds=ms:dm=mm
470 GOTO 100
490 IF df=2 THEN df=0:ds=ms:dm=mm
500 REM Monster attacks player
510 GOSUB 3300
520 IF PS<=0 THEN GOSUB 1400:GOTO 900
590 GOSUB 30:IF mf>0 THEN GOTO 250
600 IF df=2 THEN df=0:ds=ms:dm=mm
610 GOTO 100
900 REM End of game routine
999 END
1000 REM Main menu
1010 MODE 2
1020 PRINT"Choose what you want to do"
1030 PRINT"=====
1040 LOCATE 1,5:PRINT"Movement : "
1045 LOCATE 5,8:GOSUB 2600
1050 LOCATE 5,12:PRINT"Look"
1060 LOCATE 5,14:PRINT"Pause"
1070 LOCATE 5,16:PRINT"Teleport"

```

```

1080 LOCATE 1,21:PRINT"Type the first le
tter of the command of your choice"
1099 RETURN
1100 GOSUB 2500
1110 MODE 2
1120 PRINT"Choose what you want to do"
1130 PRINT"=====
1140 LOCATE 5,7:PRINT"Sword fight"
1150 LOCATE 5,9:PRINT"Dagger fight"
1160 LOCATE 5,11:PRINT"Run away"
1170 LOCATE 5,13:PRINT"Magic spell"
1175 LOCATE 1,16:GOSUB 1800
1177 LOCATE 1,18:GOSUB 2300
1180 LOCATE 1,21:PRINT"Type the first le
tter of the command of your choice"
1199 RETURN
1200 REM Player has killed the DRAGON
1210 MODE 2
1220 LOCATE 1,5:PRINT"Well played brave
knight"
1230 LOCATE 1,7:PRINT"You have slain the
foul beast"
1240 LOCATE 1,9:PRINT"The Dragon, terror
of your people, has fallen in battle"
1245 GOSUB 30:GOSUB 30
1250 LOCATE 1,12:PRINT"(. . . . and your sc
ore hath mysteriously risen . . . .)"
1260 sc=sc+200
1270 df=1
1299 RETURN
1400 REM The Player managed to get kille
d
1410 MODE 2
1420 PRINT"Now comes that moment in the
life of every truly gallant knight"
1430 PRINT:PRINT"When the court meets fo
r a great feast and all make merry and e
at
and drink their fill."
1440 GOSUB 30:PRINT:PRINT". . . .except you
that is. . . . ."
1450 PRINT:PRINT"I'm afraid that you jus
t died."
1460 PRINT:PRINT"You managed to amass th
e truly noble score of ";sc
1470 LOCATE 1,21:PRINT"Why not try diffe
rent tactics next time"

```

```

1480 GOSUB 30:GOSUB 30:GOSUB 30
1499 RETURN
1500 REM The Player escaped
1510 CLS
1520 PRINT"You've done it, against all o
dds"
1530 PRINT"You've actually escaped from
the dreadful place"
1540 IF df=0 THEN PRINT"You didn't 'Quit
e' manage to stay around long enough to
kill the Dragon.":GOTO 1560
1545 PRINT"You even managed to kill the
foul, cruel and awfully nasty Dragon on
the way out"
1557 PRINT:PRINT"That won't be altogethe
r bad for your bank balance":GOSUB 30:PR
INT:PRINT"or your score"
1558 sc=sc+400
1560 LOCATE 1,16:PRINT"YOU HAVE SCORED "
;SC;" points"
1570 PRINT"Plus a small psychological bo
nus":GOSUB 30:PRINT"You're alive"
1599 RETURN
1600 REM set RAND variable by maze numbe
r requested
1610 MODE 2
1620 LOCATE 1,11
1630 INPUT"Maze number ";n
1640 LOCATE 1,18
1650 PRINT"Write this number down"
1660 PRINT:PRINT"so that you can have th
e"
1670 PRINT:PRINT"same maze again if you
lose"
1680 RANDOMIZE n
1699 RETURN
1700 REM Choose a random monster
1705 l=3*lv:IF l>20 THEN l=20
1710 n=INT(RND(1)*l+1):m=INT(RND(1)*20+1
)
1720 DJ$=a$(M)+" "+M$(N)
1730 MS=INT(RND(1)*MS(N)+1)
1740 MM=INT(RND(1)*MM(N)+1)
1750 MC=MC(N)
1760 NS=MS+MM*4*MC/100
1799 RETURN

```

```

1800 REM Print monster statistics
1810 PRINT"The ";DJ$;
1820 PRINT" is ";
1830 IF ms<4 THEN PRINT"nearly dead":GOT
0 1899
1840 IF ms<10 THEN PRINT"badly wounded":
GOTO 1899
1850 IF ms>ps THEN PRINT"still dangerous
"
1860 IF ms<=ps THEN PRINT"still here"
1899 RETURN
2000 REM reduce MS by player fighting
2010 PRINT"You attack the monster....":G
OSUB 30
2020 IF RND(1)*wp<0.4 THEN PRINT".... bu
t miss":GOTO 2099
2030 PRINT".... and hit,"
2040 n=INT(RND(1)*ps/wp+1)
2050 IF n<=ms/4 THEN PRINT"causing a sma
ll amount of damage":GOTO 2090
2060 IF n>ms/2 THEN PRINT"causing seriou
s damage to the monster"
2070 IF n>=ms THEN PRINT:PRINT"The monst
er dies from it's wounds"
2090 ms=ms-n:IF ms<0 THEN ms=0
2095 IF ms<=0 THEN sc=sc+ns:mf=0:IF df=2
THEN GOSUB 1200
2099 RETURN
2100 REM Reduce MS by player magic
2110 PRINT"Waving your hands and chantin
g":PRINT"You begin to summon the fire de
mon"
2120 IF RND(1)*pm<mm THEN PRINT"but your
spell is disrupted by your foe":PRINT"a
nd the demon returns":PRINT"to the hells
where it be
longs":GOTO 2199
2130 PRINT"and gradually the demon forms
around your opponents body"
2135 GOSUB 30
2140 n=INT(RND(1)*pm+1)
2145 ms=ms-n:pm=pm-n
2150 IF ms<0 THEN ms=0
2160 IF ms=0 THEN PRINT"The ";DI $;" is
slain by the fire demon":LET sc=sc+ns:mf
=0:IF df=2 THEN GOSUB 1200

```

```

2199 GOSUB 30:RETURN
2200 REM Set Teleport, Heal, Attack menu
2210 MODE 2
2220 LOCATE 2,6:PRINT"SELECT TYPE OF SPE
LL"
2230 LOCATE 2,7:PRINT"=====
=="
2240 LOCATE 1,11:PRINT"T)eleport"
2250 LOCATE 1,13:PRINT"H)eal"
2260 LOCATE 1,15:PRINT"A)ttack"
2265 LOCATE 1,17:GOSUB 1800
2270 LOCATE 6,19:PRINT"Press initial let
ter to select"
2280 GOSUB 3400
2290 op=INSTR("THA",i$)
2298 IF op=0 THEN GOTO 2280
2299 RETURN
2300 REM Display player characteristics
2310 PRINT"You are feeling ";
2320 ON INT(ps/5) GOTO 2340,2350,2360
2330 PRINT"very strong":GOTO 2370
2340 PRINT"very weak":GOTO 2370
2350 PRINT"weak":GOTO 2370
2360 PRINT"fairly strong"
2370 IF pm=0 THEN PRINT"and you have no
magic left":GOTO 2399
2380 IF pm<5 THEN PRINT"You're running o
ut of magic"
2399 RETURN
2400 REM Read names and adjectives
2410 RESTORE 9000
2420 FOR l=1 TO 20
2430 READ a$(L):NEXT l
2450 FOR l=1 TO 20
2460 READ m$(L):NEXT l
2470 FOR l=1 TO 20
2480 READ ms(l),mm(l),mc(l)
2490 NEXT l
2499 RETURN
2500 REM Generate MO$ movement string
2505 IF I$="" THEN I$="F"
2510 FOR l=65 TO ASC(i$):a=RND(1):NEXT l
2520 IF RND(1)<0.3 THEN mo$="N" ELSE mo$
=" "
2530 IF RND(1)<0.3 THEN mo$=mo$+"S" ELSE
mo$=mo$+" "

```



```

2540 IF RND(1)<0.3 THEN mo$=mo$+"E" ELSE
    mo$=mo$+" "
2550 IF RND(1)<0.3 THEN MO$=MO$+"W" ELSE
    MO$=MO$+" "
2560 IF RND(1)<0.07 THEN MO$=MO$+"U" ELS
E MO$=MO$+" "
2570 IF RND(1)<0.07 THEN MO$=MO$+"D" ELS
E MO$=MO$+" "
2580 IF MO$="      " THEN GOTO 2520
2590 IF RND(1)<0.4 THEN MF=1
2599 RETURN
2600 REM DESCRIBE LOCATION EXITS
2610 PRINT"You can move : ";mo$
2699 RETURN
2700 REM Player chose the Heal spell
2710 PRINT"You being to heal yourself"
2720 l=INT(RND(1)*3+1)
2730 IF l=1 THEN PRINT"but your spell la
cks something"
2740 IF l=3 THEN PRINT"wow ! what a powe
rful healing spell"
2750 n=INT(RND(1)*pm+1)
2760 ps=ps+n*1:pm=pm-n
2770 IF pm=0 THEN PRINT"You have used up
all your magic"
2799 RETURN
2800 REM Teleport out of battle routine
2810 IF RND(1)*pm<mm*1.5 THEN PRINT>Your
teleport spell fails":GOTO 2899
2820 PRINT>Your material form shimmers a
nd then vanishes"
2830 GOSUB 30:PRINT>You re-appear somewh
ere else"
2840 RANDOMIZE TIME
2850 PRINT"The surroundings look strange
"
2860 IF RND<0.4 THEN PRINT>You've moved
further down into the dungeon":lv=lv+INT
(RND(1)*4+1)
2870 mf=0
2880 IF df=2 THEN df=0
2899 RETURN
2900 REM Setup all variables/arrays
2920 DIM a$(20),m$(20)
2930 DIM ms(20),mm(20),mc(20)
2940 GOSUB 2400

```

```

2945 dm=30:dc=75:ds=25
2950 df=0
2960 ps=20:pm=20
2970 sc=0:lv=3
2980 mf=0
2999 RETURN
3000 REM Transfer DRAGON statistics
3010 mm=dm:mc=dc:ms=ds
3040 ns=0
3050 DJ$="awe inspiring, horrendous Drag
on"
3099 RETURN
3100 REM the monster attacks (fighting)
3105 PRINT"The monster claws and bites a
t you":GOSUB 30
3110 IF RND(1)<0.4 THEN PRINT"but misses
you":GOTO 3199
3120 PRINT"and hits you"
3130 n=INT(RND(1)+ms+1)
3140 ps=ps-n:IF ps<0 THEN ps=0
3150 IF ps<5 THEN PRINT"You're getting p
retty badly hurt":PRINT"better watch you
r step"
3199 RETURN
3200 REM the monster attacks (magically)
3205 PRINT"The monster begins to summon
a demon":GOSUB 30
3210 IF RND(1)*mm<pm THEN PRINT"the demo
n of cold is banished":GOTO 3299
3220 n=INT(RND(1)*mm+1)
3230 ps=ps-n:mm=mm-n
3240 IF ps<=0 THEN ps=0:PRINT"You froze.
...."
3250 IF ps<5 THEN PRINT"You're nearly de
ad"
3260 IF ps>=5 THEN PRINT"Brrr. That was
chilly"
3299 RETURN
3300 REM Choose monster attack type
3310 IF RND(1)*ps>10*RND(1)*ms THEN PRIN
T"The monster runs away from you":mf=0:s
c=sc+ns:GOTO 3399
3320 IF RND(1)*100<mc THEN GOSUB 3200:GO
TO 3399
3330 GOSUB 3100
3399 RETURN


```

```

3400 REM Read a character from KBD:
3420 i$=INKEY$
3425 IF i$>="a" AND i$<="z" THEN i$=CHR$
(ASC(i$)-ASC("a")+ASC("A"))
3430 IF i$<"A" OR i$>"Z" THEN GOTO 3420
3499 RETURN
9000 DATA vicious,nasty,enormous,violent
9001 DATA sly,undead,beautiful,evil
9002 DATA rotting,slimy,armoured
9003 DATA terrible,foul,bloody
9004 DATA fearsome,hypnotic,dangerous
9005 DATA sadistic,cruel,pitiful
9010 DATA orc,kobold,newt,wolf
9015 DATA eagle,dwarf,troll,balrog
9020 DATA elf,wight,king,bear
9025 DATA black rider,snake,devil,will o
' the wisp
9027 DATA slime,giant,venus flower,blowf
ly
9030 DATA 10,0,0,5,20,75,1,50,100,20,0,0
9040 DATA 15,5,20,15,0,0,30,0,0,15,20,50
9050 DATA 10,40,25,40,100,2,30,10,20,35,
0,0
9060 DATA 20,40,0,10,0,0,4,20,40,20,70,1
0
9070 DATA 4,0,0,35,0,0,20,10,30,10,10,75

```

SNIPER



As you will know, your Amstrad can produce figures for use in many of your games programs. We have used him here, but not in the nicest way.

He is running for his life.

The figure will appear on the screen, sometimes for a few seconds, sometimes for a split second, and you must line up your rifle and bring him down before he vanishes again.

It's not nice, as we said, but then again he might be an advance man from a foreign planet and we all know what a nuisance they can be.

How to play

When the character appears use ARROW keys to move and the COPY key to fire.

The program

```

10 REM SNIPER
20 ENV 1,8,-1,1
30 ENT 1,8,1,1
40 ENV 2,30,-3,10,15,-2,5
50 DEFINT A-Z
60 BASE$=" "+CHR$(160)+" "
70 SPIES=0
80 HITS=0
90 SPYFLAG=0
100 SPYCH=250
110 REM NEW SNIPER LOOP
120 MY=0
130 BASE=10
140 MODE 0
150 LOCATE 1,25
160 PAPER 7:PEN 4
170 FOR Q=1 TO 20
180 PRINT CHR$(127);
190 NEXT Q:PAPER 0
200 BORDER 5
210 INK 2,7
220 PEN 11
230 LOCATE BASE-1,24:PRINT BASE$;
240 REM MAIN LOOP
250 IF MY<>0 THEN GOSUB 580
260 FLAG=0
270 IF INKEY(8)=0 AND BASE>2 THEN BASE=BASE-1:FLAG=1
280 IF INKEY(1)=0 AND BASE<19 THEN BASE=BASE+1:FLAG=1
290 IF FLAG=1 THEN PEN 11:LOCATE BASE-1,24:PRINT BASE$;
300 IF INKEY(9)=0 AND MY=0 THEN MY=23:MX=BASE:SOUND 130,0,50,15,2,0,15
310 GOSUB 410
320 IF SPIES<>10 THEN GOTO 240
330 REM ALL SPIES RUN AWAY OR SHOT
340 CLS:PRINT"ALL SPIES HAVE BEEN KILLED"
"
350 LOCATE 1,5:PRINT"YOU HIT ";HITS;" OF THEM"
360 IF HITS>8 THEN PRINT"      ( there may be a medal in this for you )"


```

```

370 IF INKEY$=" " THEN GOTO 370
380 IF INKEY$("<>") THEN GOTO 380
390 RUN
400 END
410 REM PERFORM SPY ACTIONS
420 IF SPYFLAG=1 THEN GOTO 460
430 IF RND(1)>0.007 THEN RETURN
440 SPYPOS=INT(RND(1)*10)+6:SPYDIR=1:IF
RND(1)>0.5 THEN SPYDIR=-1
450 GOTO 570
460 REM SPY IS PRESENT
470 IF SPYPOS=1 OR SPYPOS=19 THEN LOCATE
  SPYPOS+1,5:PRINT " ";GOTO 560
480 IF RND(1)>0.09 THEN GOTO 550
490 SOUND 1,0,10,3,1,1,6
500 SPYPOS=SPYPOS+SPYDIR
510 LOCATE SPYPOS,5
520 PEN 3
530 SPYCH=501-SPYCH
540 PRINT " ";CHR$(SPYCH);" ";
550 RETURN
560 PRINT " ";SPIES=SPIES+1
570 SPYFLAG=1-SPYFLAG:RETURN
580 REM MOVE MISSILE UP
590 LOCATE MX,MY:PRINT " ";
600 MY=MY-1
610 IF MY=0 THEN RETURN
620 LOCATE MX,MY
630 IF MX=SPYPOS+1 AND MY=5 THEN HITS=HI
TS+1:PRINT " ";SPYFLAG=0:MY=0:SOUND 3,2
00+20*SPIES,15:SPYPOS=0:RETURN
640 PEN 9:PRINT CHR$(148);
650 RETURN

```

WAR PLAN



Twenty enemy units are 'hidden' on the screen and must be 'knocked' out by your missiles in as few shots as possible. You have only fifty missiles so you will have to use them sparingly, and accurately, or you will be over-run by the opposition.

If, after firing, you have not hit your target you will receive intelligence reports from your forward observers in the field.

Your forward troops will only be able to 'search' eight surrounding squares after a miss but, if you make a direct hit then twenty four surrounding squares will be scanned for further enemy targets.

How to play

Use the letter and number keys to define the co-ordinates you have chosen and the ENTER key to launch your missiles.

The program

```

10 REM WAR-PLAN
20 GOSUB 380
30 SYMBOL AFTER 129
40 SYMBOL 129,0,32,96,120,96,112,136,112
50 SYMBOL 130,0,0,0,0,0,0,0,0
60 SYMBOL 131,0,62,62,62,32,32,32,32
70 SYMBOL 132,0,136,80,32,32,32,80,136
80 GOSUB 440
90 GOSUB 510
100 GOSUB 860
110 GOSUB 1010
120 GOSUB 1170
130 IF NME(X,Y)=0 THEN GOSUB 1210:GOTO 1
60
140 IF NME(X,Y)=-1 THEN GOSUB 1240:GOTO
160
150 GOSUB 1320
160 IF NME(X,Y)=0 THEN GOSUB 1460 ELSE G
OSUB 1380
170 SR=SR-1:IF SR>0 THEN M$="You have "+
STR$(SR)+" shots left":GOSUB 630:PRINT
180 IF SR>0 AND ER>0 THEN GOTO 110
190 REM PLAYER IS OUT OF SHOTS
200 REM OR HAS DESTROYED ALL NME UNITS
210 IF ER=0 THEN GOSUB 1540:GOTO 90
220 M$="Central Office advises you.":GOS
UB 630:PRINT
230 M$="You have used up all our missile
s":GOSUB 630:PRINT
240 M$="This is where the enemy was hidi
ng":GOSUB 630:PRINT
250 WINDOW 1,40,1,20:GOSUB 930:WINDOW 1,
40,21,25
260 M$="You have scored :"+STR$(sc):GOSU
B 630:PRINT:FOR L=1 TO 1000:NEXT L
270 IF SC>HS THEN M$="Which is the high
score":GOSUB 630:HS$=N$:HS=SC:PRINT
280 REM Ask IF PLAYER WANTS ANOTHER GAME
290 WINDOW 1,40,1,25:CLS:LOCATE 1,11
300 M$="Commander "+HS$+" has the high s
core":GOSUB 630

```



```

310 LOCATE 1,13:M$="Which is "+STR$(hs)+
" points":GOSUB 630
320 LOCATE 1,17:LET M$="Does anyone want
another game (Y/N) ?":GOSUB 630
330 I$=INKEY$:IF I$<>" " THEN GOTO 330
340 I$=INKEY$:IF I$="" THEN GOTO 340
350 IF I$="y" OR I$="Y" THEN GOTO 80
360 IF I$="n" OR I$="N" THEN WINDOW 1,40
,1,25:CLS:END
370 GOTO 330
380 REM SET UP FOR A NEW GAME
390 HS=0:HS$=""
400 MODE 1
410 BORDER 13
420 DIM NME(14,24)
430 RETURN
440 REM SET UP FOR A NEW PLAYER
450 WINDOW 1,40,1,25
460 SC=0:CLS
470 LOCATE 6,11:PRINT"What is your name
";
480 INPUT n$:IF N$="" THEN GOTO 470
490 WV=1
500 RETURN
510 REM SET UP FOR NEW WAVE
520 WINDOW 1,40,1,25:CLS:LOCATE 1,9:M$="
Commander "+N$+" ":GOSUB 630
530 LOCATE 1,11:M$="Central Office advis
es you.":GOSUB 630
540 LOCATE 1,13:M$="Attack enemy immedia
tely.":GOSUB 630
550 LOCATE 1,15:M$="Repeat, ATTACK. ATTA
CK.":GOSUB 630
560 LOCATE 1,21:M$="":FOR L=1 TO 10:M$=M
$+CHR$(10):NEXT L:GOSUB 630
570 FOR X=1 TO 14:FOR Y=1 TO 24:NME(X,Y)
=0:NEXT Y:NEXT X
580 ER=20:FOR L=1 TO ER
590 X=INT(RND(1)*10+3):Y=INT(RND(1)*20+3
):IF NME(X,Y)<>0 THEN GOTO 590
600 IF L<3 THEN NME(X,Y)=-1 ELSE NME(X,Y
)=1
610 NEXT L:SR=50
620 RETURN
630 REM Teletype emulator
640 FOR L=1 TO LEN(M$)

```

```

650 PRINT MID$(M$,L,1);
660 SOUND 1,0,5,7,0,0,10:SOUND 1,0,3
670 IF SQ(1)>127 THEN GOTO 670
680 NEXT L
690 RETURN
700 REM SCAN LOCALLY
710 NN=0
720 FOR X=TX-1 TO TX+1
730 FOR Y=TY-1 TO TY+1
740 NN=NN+ABS(NME(X,Y))
750 NEXT Y:NEXT X
760 NN=NN-ABS(NME(TX,TY))
770 RETURN
780 REM SCAN LARGER AREA
790 NN=0
800 FOR X=TX-2 TO TX+2
810 FOR Y=TY-2 TO TY+2
820 NN=NN+ABS(NME(X,Y))
830 NEXT Y:NEXT X
840 NN=NN-ABS(NME(TX,TY))
850 RETURN
860 REM LABEL MAP AREA
870 WINDOW 1,40,1,20:CLS
880 LOCATE 12,1:PRINT"11111111112":LOCAT
E 3,2:PRINT"12345678901234567890"
890 LOCATE 1,3
900 FOR L=ASC("A") TO ASC("J")
910 PRINT CHR$(L):NEXT L
920 RETURN
930 REM FILL IN ALL NME UNITS
940 FOR X=3 TO 12
950 LOCATE 3,X
960 FOR Y=3 TO 22
970 PRINT CHR$(130+NME(X,Y));
980 NEXT Y
990 NEXT X
1000 RETURN
1010 REM GET A LEGAL SHOT
1020 WINDOW 1,40,21,25:M$="Enter target
for missile ":GOSUB 630:PRINT
1030 X=1:A=1:B=0:C=1:WINDOW 1,40,1,20
1040 LOCATE 11,15:PRINT"[ ";CHR$(64+X);
";A+B;" ] ";
1050 I$=INKEY$:IF I$<>"" THEN GOTO 1050
1060 I$=INKEY$:IF I$="" THEN GOTO 1060
1070 IF I$<"0" OR I$>"9" THEN GOTO 1100

```

```

1080 IF C=1 THEN B=A*10:A=ASC(I$)-ASC("O
"):C=0:GOTO 1040
1090 C=1:B=A*10:A=ASC(I$)-ASC("O"):GOTO
1040
1100 IF I$=CHR$(13) THEN GOTO 1140
1110 IF I$>="a" AND I$<="j" THEN I$=CHR$
(ASC(I$)-ASC("a")+ASC("A"))
1120 IF I$<"A" OR I$>"J" THEN PRINT CHR$
(7);:GOTO 1040
1130 X=ASC(I$)-ASC("A")+1:GOTO 1040
1140 Y=A+B+2:X=X+2:IF X=2 OR Y<3 OR Y>22
THEN A=1:B=0:X=0:PRINT CHR$(7);:GOTO 10
30
1150 LOCATE 11,15:PRINT"
";:WINDOW 1,40,21,25
1160 RETURN
1170 REM UPDATE MAP
1180 WINDOW 1,40,1,20:LOCATE Y,X:PRINT C
HR$(132);
1190 WINDOW 1,40,21,25
1200 RETURN
1210 REM PLAYER MISSED
1220 M$="The missile missed the enemy":G
OSUB 630:PRINT
1230 RETURN
1240 REM PLAYER HIT A COMMAND BASE
1250 M$="You have hit the command base":
GOSUB 630:PRINT
1260 IF er>2 THEN m$="...but it is not d
estroyed":GOSUB 630:PRINT:GOTO 1310
1270 M$="...and you destroyed it":GOSUB
630:PRINT
1280 SC=SC+500
1290 NME(X,Y)=0
1300 ER=ER-1
1310 RETURN
1320 REM PLAYER HIT AN ENEMY UNIT
1330 NME(X,Y)=0
1340 M$="You have hit an enemy unit":GOS
UB 630:PRINT
1350 M$="and it is destroyed":GOSUB 630:
PRINT:ER=ER-1
1360 SC=SC+20
1370 RETURN
1380 REM SEARCH AROUND NEARBY


```

```

1390 M$="Intelligence reports :
"+CHR$(13)+CHR$(10)
1400 tx=x:TY=Y:GOSUB 700
1410 IF NN=0 THEN M$=M$+"no nearby activ
ity"
1420 IF nn>0 AND nn<3 THEN m$=m$+"A smal
l amount of nearby activity"
1430 IF nn>=3 THEN m$=m$+"A large amount
of nearby activity"
1440 WINDOW 1,40,21,25:GOSUB 630:PRINT
1450 RETURN
1460 REM search around after a hit
1470 m$="Intelligence reports :
"+CHR$(13)+CHR$(10)
1480 tx=x:ty=y:GOSUB 780
1490 IF nn=0 THEN M$=M$+"no local activi
ty"
1500 IF nn>0 AND nn<6 THEN m$=m$+"a smal
l amount of local activity"
1510 IF nn>5 THEN m$=m$+"a large amount
of local activity"
1520 WINDOW 1,40,21,25:GOSUB 630:PRINT
1530 RETURN
1540 REM PLAYER HAS WIPED OUT A WAVE
1550 WINDOW 1,40,1,25:CLS
1560 LOCATE 1,11:M$="Cental Office advis
es you.":GOSUB 630
1570 LOCATE 1,13:m$="Forthcoming enemy c
ounterattack.":GOSUB 630
1580 LOCATE 1,15:m$="Prevent at all cost
s.":GOSUB 630
1590 LOCATE 1,19:m$="Your score so far i
s :"+STR$(sc)+" points.":GOSUB 630
1600 LOCATE 1,21:m$="":FOR l=1 TO 10:m$=
m$+CHR$(10):NEXT l:GOSUB 630
1610 RETURN

```

S.I.M.P.E.R.



At last, you say, in a computer book there must be an acronym somewhere because the whole business is full of Poke, Peek, Byte and Ram. Well, our title stands for Semi Intelligent Machine Psychologist with Extendable Response.

What, you ask, has that got to do with the time of day?

We will explain.

This is your chance to use your computer to make fools out of all your friends, party guests, loved? ones etc. Your SIMPERING Amstrad will ask questions and respond to your answers with 'semi-intelligent' comments. Can you take the risk?

How to play

Your computer will ask you to enter statements about yourself.

Type in your comments, making them as long as possible, and your Amstrad will make it's considered comment on your character.

It's a little like having a 'shrink' at home.

Programming Hints

We have given you a selection of appropriate data (words) that might make fun for your family but you can change, add or delete these lines and cause fun and games at your next party.

If you have added a memory expansion unit to your computer you can add to the data statements from lines 520 onwards and give the computer a much wider choice of response.

The Program

```

10 REM S.I.M.P.E.R.
20 Q$="Hello. My name is SIMPER, I'm her
e to discuss your problems. Why do you e
njoy life ?"
30 bs=0
40 os=0
50 PRINT q$:PRINT
60 INPUT a$:IF a$="" THEN GOTO 50
70 IF LEFT$(a$,3)="YES" THEN PRINT"At la
st, some determination":q$="Do you often
behave quite so firmly ?":GOTO 50
80 IF LEFT$(a$,6)="PARDON" OR LEFT$(A$,2
)="EH" OR A$="WHAT" THEN GOTO 50
90 IF A$="NO" THEN PRINT"Short and to th
e point":IF OS=0 THEN PRINT"Give me some
thing to go on":GOTO 50
100 GOSUB 250:GOSUB 300
110 A$=" "+A$
120 IF V=0 AND U<>0 THEN GOSUB 360:GOTO
50
130 IF V=0 AND OS<>0 THEN U=OS:OS=0:GOSU
B 360:GOTO 50

```

```

140 IF V=0 THEN GOSUB 460:GOTO 50
150 GOSUB 410:OS=U
160 GOTO 50
170 REM SEARCH FOR T$ IN A$
180 F=0
190 IF A$=T$ THEN F=1:RETURN
200 IF LEN(A$)<LEN(T$) THEN RETURN
210 FOR L=1 TO LEN(A$)-LEN(T$)+1
220 IF MID$(A$,L,LEN(T$))=T$ THEN F=1
230 NEXT L
240 RETURN
250 REM RETURN VERB NUMBER IN V
260 V=0:RESTORE 520
270 READ N:READ T$:FOR Q=1 TO N
280 READ T$:GOSUB 170:IF F=1 THEN V=Q
290 NEXT Q:RETURN
300 REM RETURN NOUN NUMBER IN U
310 U=0:RESTORE 550
320 READ N:READ T$:FOR Q=1 TO N
330 READ T$:GOSUB 170:IF F=1 THEN U=Q
340 NEXT Q
350 RETURN
360 REM FIND REPLY SENTANCE
370 RESTORE 790
380 FOR L=0 TO (U-1)*3+INT(RND(1)*3)
390 READ Q$:NEXT L
400 RETURN
410 REM CHOOSE SENTANCE
420 RESTORE 590
430 FOR L=0 TO (V-1)*3+INT(RND(1)*3)
440 READ Q$:NEXT L
450 RETURN
460 REM CONJURE UP A RESPONSE
470 BS=BS+1:IF BS=7 THEN PRINT"SORRY ...
  YOUR TIME IS UP":PRINT:PRINT:PRINT:PRIN
T:END
480 RESTORE 980
490 FOR L=1 TO BS
500 READ Q$:NEXT L
510 RETURN
520 DATA 8
530 DATA !
540 DATA LOVE,HATE,OWN,WANT,FLY,WALK,TAK
E,GIVE
550 DATA 9
560 DATA !

```

570 DATA CAR, COMPUTER, HOLIDAY, MONEY, TELE
 , MUSIC, VIDEO, FOOD, FISH
 580 REM REPLY DATA
 590 DATA How do you know that it's love
 ?
 600 DATA What does it feel like to be in
 love ?
 610 DATA Why do you think you feel like
 that ?
 620 DATA Are you sure that you're not be
 ing over hasty ?
 630 DATA Well if you're sure about it th
 en it's good for you
 640 DATA Is this feeling very intense. D
 o you really mean it ?
 650 DATA That's very progressive of you,
 wouldn't 'CARE FOR' be a better way of
 putting it ?
 660 DATA But do you own your soul. or ha
 ve you traded it with consumer society ?
 670 DATA Please clarify yourself. Love o
 r want. The first is more polite.
 680 DATA That's very impulsive
 690 DATA I Want....I Want... Belloc has
 a suprise for you
 700 DATA Do you sometimes wish that you
 could fly, Aeroplanes are so expensive.
 aren't they ?
 710 DATA Fly me to the moon...Do you thi
 nk that you'd enjoy that ?
 720 DATA Nothing beats shank's pony, wal
 king is very good exercise
 730 DATA It might be quicker by car
 740 DATA Perhaps if you said please firs
 t it might help, Your'e showing that pro
 gressive trait again
 750 DATA How would you take that ?
 760 DATA Giving is good for the characte
 r. Do you do a lot of it ?
 770 DATA Christmas is a time for giving
 780 DATA Do you recieve more than you gi
 ve ? Be honest.
 790 DATA Do you like fast cars, are you
 good mechanically, do you like motorways.
 They're too fast for me
 800 DATA You obviously like computers. I

can tell that you are a natural at the keyboard.

810 DATA Do computers worry you in any way ? ,Isn't it a bit unnerving talking to a computer like this ?

820 DATA Would you please take me with you on your holiday. That's a great way to get away from it all

830 DATA Do you like to go abroad. or perhaps you prefer to stay at home

840 DATA Money is everyone's favourite problem. Talking of money I have these rather large fees....

850 DATA How do you earn you money. Are you by any chance a fortune teller ?

860 DATA What it must be to be a millionaire. All that caviar. It's enough to make anyone ill.

870 DATA I'm worried about the profusion of tele-this and tele-that devices.

880 DATA Where do you watch television most often

890 DATA Are you addicted to television. Like some people I know

900 DATA what type of music do you like best, I used to adore Simon and Garfunkle. What do you think of them ?

910 DATA So much modern music is too loud. I get lots of cases who are disco-goers.

920 DATA Do you have a video recorder at home ? ,how long hve you had a video ?

930 DATA If you video too much then you won't have any time to notice the real world

940 DATA So you eat too much do you, food is all very well but you musn't be obsessed with it. You'll get fat.

950 DATA What's your favourite meal ?

960 DATA Would you like to talk about fish, how about some nice fish and chips

970 DATA Have you ever spent a day fishing by the river and not caught anything ?

980 DATA Tell me more about your talents . Do you play the concert piano ?

990 DATA What kind of hobbies have you got.


1000 DATA Where do you work ? or do you still go to school or college ?

1010 DATA How about a bit of word association. +++ MATHS +++ ?

1020 DATA What does the word 'MACKEREL' suggest to you ?

1030 DATA I'm afraid that I will have to leave you in a few moments. I'm expecting a very important client.

WORD SQUARE



This program is designed to test the wits of members of your circle who think they are keen of eye and sharp of brain.

The screen will generate a square of alphabet which will contain hidden words. The words can be written horizontally, vertically, diagonally and even backwards.

How to play

When you have identified a word type in the answer and press ENTER. You should note that you may identify small words such as 'of' and 'for' which appear simply because of the structure of the screen, but are not actually acceptable to the computer.

Programming Hints

To make the program challenging you should have someone else enter the data words which are in the end of the program. Otherwise you will, of course, know the words you are seeking.

The Program

```

10 REM WORDSQUARE
20 MODE 1
30 BORDER 13
40 GOSUB 500
50 DIM F(10),X(10),Y(10)
60 DIM S$(255),T$(255)
70 DIM W(10)
80 GOSUB 680
90 GOSUB 1030
100 FOR L=1 TO 10:W(L)=0
110 NEXT L
120 SC=0
130 GOTO 160
140 LOCATE 1,22:PRINT"BE SENSIBLE"
150 FOR L=1 TO 1500:NEXT L
160 LOCATE 1,23:PRINT"Your Guess
"
170 LOCATE 12,23
180 INPUT A$
190 IF LEN(A$)=0 OR LEN(A$)>10 THEN GOTO
140
200 LOCATE 1,22
210 PRINT"
"
220 GOSUB 1180
230 IF F=0 THEN LOCATE 1,22:PRINT"Not th
ere":GOTO 400
240 SC=SC+1
250 REM HIGHLIGHT THE WORD WHICH HAS BEE
N FOUND
260 IF W(F)=1 THEN LOCATE 1,22:PRINT"You
've already had that word":GOTO 400
270 W(F)=1
280 B=INT(P(F)/15)

```

```

290 A=P(F)-15*B
300 FOR L=1 TO 0
310 LOCATE A+6,B+3
320 GOSUB 640:D$=CHR$(DD)
330 LOCATE A+6,B+3:PEN 2
340 PRINT D$;
350 PEN 1
360 A=A+X(F):B=B+Y(F)
370 NEXT L
380 IF SC=10 THEN GOTO 430
390 LOCATE 1,23:PRINT"Well done ";SC;" c
orrect so far"
400 FOR L=1 TO 2000:NEXT L
410 LOCATE 1,23:PRINT"
"
420 GOTO 160
430 REM Here when all ten are found
440 CLS
450 LOCATE 1,11
460 PRINT"Well done...you found them all
"
470 IF INKEY$<>"" THEN GOTO 470
480 IF INKEY$="" THEN GOTO 480
490 END
500 REM PUT MACHINE CODE JUST ABOVE RE-A
DJUSTED HIMEM
510 MEMORY 39000
520 MC=HIMEM+1
530 TA=HIMEM+&7F
540 TH=INT(TA/256)
550 TL=TA-256*TH
560 POKE MC,&CD
570 POKE MC+1,&60
580 POKE MC+2,&BB
590 POKE MC+3,&32
600 POKE MC+4,TL
610 POKE MC+5,TH
620 POKE MC+6,&C9
630 RETURN
640 CALL MC
660 DD=PEEK(TA)
670 RETURN
680 REM READ IN DATA AND STORE IN SQUARE
690 LOCATE 11,11:PRINT"ONE MOMENT"
700 S$=""
710 FOR L=1 TO 15

```


```

720 S$=S$+"
730 NEXT L
740 RESTORE 1270
750 FOR I=1 TO 10
760 READ N$
770 N=LEN(N$)
780 N$=RIGHT$(N$,N)
790 ON INT(RND*8) GOSUB 1100,1110,1120,1
130,1140,1150,1160,1170
800 A=INT(RND(1)*15):B=INT(RND(1)*15)
810 C=A+X(I)*N
820 D=B+Y(I)*N
830 IF C<0 OR D<0 OR C>14 OR D>14 THEN G
OTO 790
840 L=-1
850 L=L+1
860 Q=1+(B+Y(I)*L)*15+A+X(I)*L
870 Q$=MID$(S$,Q,1)
880 IF Q$=" " THEN GOTO 900
890 IF Q$<>MID$(N$,L+1,1) THEN GOTO 790
900 IF L<>N-1 THEN GOTO 850
910 FOR L=0 TO N-1
920 Q=1+(B+Y(I)*L)*15+A+X(I)*L
930 S$=LEFT$(S$,Q-1)+MID$(N$,L+1,1)+MID$
(S$,Q+1)
940 NEXT L
950 P(I)=A+B*15
960 NEXT I
970 T$=""
980 FOR L=1 TO 255
990 IF MID$(S$,L,1)=" " THEN T$=T$+CHR$(
INT(RND(1)*26+65)) ELSE T$=T$+MID$(S$,L,
1)
1000 NEXT L
1010 S$=T$:T$=""
1020 RETURN
1030 REM DISPLAY SQUARE
1040 CLS
1050 FOR L=0 TO 14
1060 LOCATE 6,L+3
1070 PRINT MID$(S$,L*15+1,15)
1080 NEXT L
1090 RETURN
1100 X(I)=-1:Y(I)=-1:RETURN
1110 X(I)=1:Y(I)=1:RETURN
1120 X(I)=-1:Y(I)=1:RETURN

```

```
1130 X(I)=1:Y(I)=-1:RETURN
1140 X(I)=1:Y(I)=0:RETURN
1150 X(I)=0:Y(I)=1:RETURN
1160 X(I)=-1:Y(I)=0:RETURN
1170 X(I)=0:Y(I)=-1:RETURN
1180 REM CHECK IF RESPONSE IS VALID
1190 RESTORE 1270
1200 F=0
1210 FOR L=1 TO 10
1220 READ N$
1230 IF N$=A$ THEN F=L:O=LEN(N$)
1240 NEXT L
1250 RETURN
1260 REM WORD LISTS
1270 DATA AMSTRAD,VIDEO,COLOUR
1280 DATA SOUND,GRAPHICS,GAMES
1290 DATA MONTIOR,PLAY,COMPUTER
1300 DATA WATER
```

FAST TYPER



How are your reactions, your spelling and your keyboard skills? On your screen you will see a fast moving, ever changing colour background moving from left to right across the line of your vision.

In the middle of this screen will pop up the word which you have to copy type on the screen to stop the computer.

How to play

You simply type in the word, without pressing any Enter keys, in as quick a time as possible to achieve the highest score.

Programming hints

If you find the words in lines 450 and 460 too simple, or even too difficult, you can change them to suit your own needs. All you need to do is to enter your own data, remembering to change line 440 if you increase the number of words from 8.

If you wish to change the length of the words then remember to change the number 6 in line 170 to the appropriate length of the new words you wish to insert.

The program


```

10 REM FAST TYPER
20 GOSUB 330
30 GOES=0
40 SC=0
50 EVERY 1,1 GOSUB 270
60 RESTORE:READ N
70 FOR L=1 TO INT(RND(1)*N+1):READ W$
80 NEXT L
90 IF RND(1)>0.001 THEN GOTO 90
100 INK 12,26:PEN 12
110 LOCATE 8,10:P=1
120 T=0
130 PRINT W$
140 T=T+1:IF INKEY$<>" " THEN GOTO 140
150 T=T+1:A$=INKEY$:IF A$="" THEN GOTO 1
50
160 IF A$<>MID$(W$,P,1) THEN GOTO 140
170 P=P+1:IF P<>6 THEN GOTO 140
180 LOCATE 8,10:PRINT "      ";
190 SC=SC+T:GOES=GOES+1:IF GOES<>5 THEN
GOTO 60
200 SC=INT(SC/5)
210 CLS:LL=REMAIN(1)
220 LOCATE 5,10:PRINT"AVERAGE TIME :";SC
230 LOCATE 3,15:PRINT"<SPACE> TO TRY AGA
IN"
240 IF INKEY$=" " THEN GOTO 240
250 IF INKEY$<>" " THEN GOTO 250
260 RUN
270 REM LINE SLIDING
280 REM USING 10 COLOURS
290 DI
300 ZZ=LL:LL=LL+1:IF LL>10 THEN LL=1
310 INK LL,14:INK ZZ,0
320 EI:RETURN
330 REM PLOT LINES
340 MODE 0

```

```
350 FOR L=1 TO 10:INK L,L*2:NEXT L
360 CC=1
370 FOR H=1 TO 639 STEP 4
380 PLOT H,1,CC:DRAW H,399,CC
390 CC=CC+1:IF CC>10 THEN CC=1
400 NEXT H
410 BORDER 4
420 FOR LL=10 TO 1 STEP -1:INK LL,0:NEXT
  LL
430 RETURN
440 DATA 8
450 DATA MOUSE,HOUSE,WHICH,WITCH
460 DATA MAINS,PLAIN,PLANE,MICRO
```

CUBIST CONUNDRUMS



Well, to tell the truth, we had to look it up to see what it was about and it seems there was a Dutch painter named Mondrian who painted abstract art. What has that to do with your Amstrad you may ask.

Well, the fact is that your modest little computer has the talent to originate for your eyes only – abstract art. If you have had a hard day at the office, in the kitchen or at school just enter this short program into your computer – sit back and watch the results.

It really is the most beautifully relaxing experience.

Programming hints

This small but very attractive program can be loaded into the middle of one of your own programs where you are using a pause during a search in an adventure. You could also use this when a large program was loading a second part to memory.

The program

```
10 REM CUBIST CONUNDRUMS
20 MODE 0
30 BORDER 3
40 PAPER INT(RND(1)*16)
50 A=INT(RND(1)*20+1)
60 B=INT(RND(1)*(21-A)+A)
70 C=INT(RND(1)*25+1)
80 D=INT(RND(1)*(26-C)+C)
90 WINDOW A,B,C,D
100 CLS
110 IF INKEY(47)=0 THEN GOTO 110
120 IF RND(1)<0.05 THEN GOTO 40
130 INK INT(RND(1)*16),INT(RND(1)*25)
140 GOTO 40
```

WIPEOFF



I am afraid that a bunch of horrible vandals have got into your Amstrad and covered the screen with paint from an aerosol. You have been stuck with the job of cleaning up the mess before the law arrive and start to look for a likely suspect.

How to play

You have only 30 seconds to wipe off the mess using the arrow keys and if you fail to finish in time you will get a percentage score against your efforts.

Programming hints

If you are too quick at finishing the clean up operation and you would like even less time to complete then alter the Count 10 00 in line 50.

The program


```

10 REM WIPEOFF
20 GOSUB 410
30 GOSUB 260
40 SCORE=0
50 COUNT=1000
60 X=10:Y=10
70 SOUND 1,500,10,4
80 LOCATE X,Y:PRINT " ";
90 IF INKEY(8)=0 AND X>1 THEN X=X-1
100 IF INKEY(1)=0 AND X<20 THEN X=X+1
110 IF INKEY(0)=0 AND Y>1 THEN Y=Y-1
120 IF INKEY(2)=0 AND Y<25 THEN Y=Y+1
130 GOSUB 550
140 IF DD<>32 THEN SCORE=SCORE+1
150 COUNT=COUNT-1
160 IF COUNT<>0 AND NUM<>SCORE THEN GOTO
    80
170 MODE 2
180 LOCATE 1,10
190 PRINT"You have cleared ";
200 PRINT USING "###.##";SCORE*100/NUM;
210 PRINT" % of the mess"
220 LOCATE 1,15
230 PRINT"PRESS <SPACE> TO TRY AGAIN"
240 IF INKEY$<>" " THEN GOTO 240
250 GOTO 30
260 REM SPRAY AEROSOL
270 MODE 0
280 NUM=-1
290 X=10:Y=10
300 PEN 11
310 GOSUB 550
320 IF DD=32 THEN NUM=NUM+1
330 LOCATE X,Y
340 PRINT CHR$(143);
350 IF RND(1)>0.6 THEN X=X+1
360 IF RND(1)>0.6 THEN X=X-1
370 IF RND(1)>0.6 THEN Y=Y+1
380 IF RND(1)>0.6 THEN Y=Y-1
390 IF X>0 AND X<21 AND Y>0 AND Y<26 THE
N GOTO 310
400 RETURN

```

```
410 REM PUT MACHINE CODE IN MEMORY ABOVE  
    RE-ADJUSTED HIMEM  
420 MEMORY FRE(0)-&80  
430 MC=HIMEM+1  
440 TA=HIMEM+&7F  
450 TH=INT(TA/256)  
460 TL=TA-256*TH  
470 POKE MC,&CD  
480 POKE MC+1,&60  
490 POKE MC+2,&BB  
500 POKE MC+3,&32  
510 POKE MC+4,TL  
520 POKE MC+5,TH  
530 POKE MC+6,&C9  
540 RETURN  
550 LOCATE X,Y  
560 CALL MC  
570 DD=PEEK(TA)  
580 RETURN
```

FIRST ONE



This game could mean the possible destruction of your computer unless you trust all your friends to be gentle.

The game is for up to NINE players, and the object is to be the first one to press a key and stop the game.

A coloured square will appear on the screen, and then everyone must try to be first to react.

How to play

The computer will ask "How many players are there?"

Tell the computer, which will then ask for the player's names.

Player One uses key A, player Two uses key B and so on.

If you cheat and move early however, the computer will tell everyone "Guess who cheated" "It was"

The winner will be identified and their reaction time will be shown.

The program

```

10 REM FIRST ONE
20 MODE 2
30 BORDER 4
40 INPUT "How many players are there ";N
50 IF N<1 OR N>9 THEN PRINT:PRINT"I'm af
raid that you can only have from 1 to 9
players":PRINT:PRINT:GOTO 40
60 DIM n$(N):FOR L=1 TO N
70 PRINT "Name of player ";L;" :";:INPUT
n$(L)
80 IF n$(L)="" THEN GOTO 70
90 NEXT L
100 MODE 0
110 IF RND(1)>0.997 THEN GOTO 210
120 k$=INKEY$:IF k$="" THEN GOTO 110
130 FOR L=50 TO 600 STEP 5
140 SOUND 1,L,5:NEXT L
150 S=ASC(k$)-ASC("@"):IF S<1 OR S>N THE
N PRINT "ILLEGAL KEY":FOR L=1 TO 2000:NEX
T L:GOTO 100
160 MODE 2
170 LOCATE 1,8
180 PRINT "HEY EVERYONE... GUESS WHO JUST
CHEATED "
190 PRINT "It was .....";n$(S)
200 FOR L=1 TO 2000:NEXT L:GOTO 100
210 REM MAKE BLOCK APPEAR
220 LOCATE 6,10:PRINT "PRESS NOW"
230 C=0
240 C=C+1:k$=INKEY$:IF k$="" THEN GOTO 2
40
250 S=ASC(k$)-ASC("@")
260 IF S<1 OR S>N THEN GOTO 150
270 MODE 2
280 LOCATE 1,10
290 PRINT N$(S);" wins"
300 LOCATE 1,15

```

```
310 PRINT "With a reaction count of : ";  
C  
320 FOR Q=1 TO 4000:NEXT Q  
330 GOTO 100
```

WEIGHTLIFTER



Well here at last is your chance to prove that you should have been selected for the last Olympics – as a weightlifter?

Your friends will probably tell you that you haven't got the strength to lift yourself from your chair but now you can prove them wrong as you snatch, press and lift with the best of them.

On the screen you will see a man, his weights and speed tester – see how you do on your own.

How to play

Just like the real thing the lift must be completed in four different moves and each one must be properly executed or a foul sign will appear and you will have to begin again.

At the beginning of each movement a small star will move swiftly across the screen from left to right and you must stop it between two points to begin the lift.

To stop the star press the SPACE bar.

Another little problem is that the better you become at lifting the weights the faster the star will move until – we believe – you won't be able to help your alter ego win a medal but we could be wrong.

The program

```

10 REM WEIGHTLIFTER
20 GOSUB 2000
30 SC=0:WEIGHT=100
40 F=20
100 MODE 1:N=1
110 GOSUB 3000
120 GOSUB 1000
130 LOCATE 1,20:PRINT"SCORE : ";SC
140 LOCATE 20,4:PRINT"NOW LIFTING : ";WEI
GHT
145 IF INKEY$<>"" THEN GOTO 500
150 IF RND>0.008 THEN GOTO 150
160 Q=0
170 Q=Q+1:LOCATE Q,14
180 PRINT"*";
185 FOR T=1 TO F:NEXT T
190 IF INKEY$<>"" THEN GOTO 400
200 IF Q<40 THEN GOTO 170
210 GOTO 500
400 IF Q<15 OR Q>22 THEN GOTO 500
410 IF N<4 THEN N=N+1:SC=SC+WEIGHT*N/10:
LOCATE 1,14:PRINT"
";GOTO 110
420 LOCATE 25,6:PRINT"LIFTED !!!"
430 FOR B=1 TO 26
440 SOUND 1,B*20,10:BORDER B:NEXT B
450 WEIGHT=WEIGHT+10
460 IF F>1 THEN F=F-1
470 GOTO 100

```

```

500 REM FOUL LIFT
510 LOCATE 25,6:PRINT"FOUL"
520 FOR W=1 TO 200:NEXT W
530 LOCATE 1,23:PRINT"YOU MADE IT TO ";W
    EIGHT;" POUNDS"
540 IF INKEY$=" " THEN GOTO 540
550 IF INKEY$<>" " THEN GOTO 550
560 RUN
900 REM CLEAR WINDOW
910 WINDOW 1,10,1,10
920 CLS
930 WINDOW 1,40,1,25
990 RETURN
1000 REM DRAW LIFTER IN POSITION (N)
1010 GOSUB 900
1020 ON N GOTO 1100,1200,1300,1400

1100 PLOT 64,379:DRAW 64,349:DRAW 94,349
    :DRAW 94,379:DRAW 64,379
1110 PLOT 79,349:DRAW 79,300:DRAW 39,240
    :PLOT 79,300:DRAW 119,240
1120 PLOT 39,312:DRAW 39,320:DRAW 79,335
    :DRAW 119,320:DRAW 119,312
1130 PLOT 16,265:DRAW 142,265
1140 PLOT 17,240:DRAW 17,290:PLOT 18,240
    :DRAW 18,290:PLOT 19,240:DRAW 19,290
1150 PLOT 139,240:DRAW 139,290:PLOT 140,
    240:DRAW 140,290:PLOT 141,240:DRAW 141,2
    90
1199 RETURN
1200 PLOT 64,364:DRAW 64,334:DRAW 94,334
    :DRAW 94,364:DRAW 64,364
1210 PLOT 79,334:DRAW 79,290
1220 PLOT 39,240:DRAW 64,290:DRAW 94,290
    :DRAW 119,240
1230 PLOT 39,265:DRAW 39,273:DRAW 79,320
    :DRAW 119,273:DRAW 119,265
1240 PLOT 16,265:DRAW 142,265
1250 PLOT 17,240:DRAW 17,290:PLOT 18,240
    :DRAW 18,290:PLOT 19,240:DRAW 19,290
1260 PLOT 139,240:DRAW 139,290:PLOT 140,
    240:DRAW 140,290:PLOT 141,240:DRAW 141,2
    90
1299 RETURN
1300 PLOT 64,379:DRAW 64,349:DRAW 94,349
    :DRAW 94,379:DRAW 64,379

```


```

1310 PLOT 79,349:DRAW 79,300:DRAW 39,240
:PLOT 79,300:DRAW 119,240
1320 PLOT 52,341:DRAW 52,335:DRAW 60,312
:DRAW 79,335:DRAW 98,312:DRAW 106,335:DR
AW 106,341
1330 PLOT 16,341:DRAW 142,341
1340 PLOT 17,316:DRAW 17,367:PLOT 18,316
:DRAW 18,367:PLOT 19,316:DRAW 19,367
1350 PLOT 139,316:DRAW 139,367:PLOT 140,
316:DRAW 140,367:PLOT 141,316:DRAW 141,3
67
1399 RETURN
1400 PLOT 64,379:DRAW 64,349:DRAW 94,349
:DRAW 94,379:DRAW 64,379
1410 PLOT 79,349:DRAW 79,300:DRAW 39,240
:PLOT 79,300:DRAW 119,240
1420 PLOT 52,379:DRAW 64,335:DRAW 94,335
:DRAW 106,379
1430 PLOT 17,399:DRAW 17,367:DRAW 18,367
:DRAW 18,399:DRAW 19,399:DRAW 19,367
1440 PLOT 139,399:DRAW 139,367:DRAW 140,
367:DRAW 140,399:DRAW 141,399:DRAW 141,3
67
1450 PLOT 16,379:DRAW 142,379
1499 RETURN
2000 REM INSTRUCTIONS SUBROUTINE
2010 MODE 1
2020 LOCATE 10,1:PRINT"WEIGHTLIFTER"
2030 LOCATE 9,2:PRINT"===== "
2040 LOCATE 1,5:PRINT"In this game you t
ry to help Feeble Fredas he attempts to
lift weights."
2050 PRINT:PRINT"Unfortunately, his timi
ng isn't all it could be, and so you ne
ed to tell him when he must move to th
e next stage
of the lift."
2060 PRINT:PRINT"As the timing light fli
es across the screen, you must shout a
t Fred to lift while it is between the
two critica
l points."
2070 LOCATE 13,20:PRINT"< SPACE >"
2080 IF INKEY$=" " THEN GOTO 2080
2090 IF INKEY$("<>") THEN GOTO 2090
2099 RETURN

```

```
3000 REM mark critical points
3010 INK 2,11
3020 PEN 2
3030 PLOT 1,192:DRAW 639,192:PLOT 1,174:
DRAW 639,174
3040 LOCATE 15,13:PRINT CHR$(214)
3050 LOCATE 22,13:PRINT CHR$(215)
3060 LOCATE 15,15:PRINT CHR$(213)
3070 LOCATE 22,15:PRINT CHR$(212)
3090 INK 1,24:PEN 1:BORDER 5
3099 RETURN
```

KALEIDOSCOPE



Once upon a time, before the advent of sophisticated computers, we used to have little cardboard tubes with glass panels and pieces of coloured card inside. We used to shake them and peer into a little hole in the end and see a wonderful collection of shapes, colours and sizes all in lovely symmetry.

Well, enough of all this nostalgia. This is the age of the computer so here is an up to date version of an old Victorian parlour toy. We have included this program because it can be typed in by the youngest member of the family in ten minutes and they will then have the immense satisfaction of having 'entered' and 'run' their own computer program with beautifully colourful results.

Programming hint

This is a handy little program for entering into a main program as a colourful sub-routine. Simply enter a loop sub-routine and insert this listing.

The program

```
10 REM KALEIDESCOPE
20 K$=CHR$(207)
30 DEFINT A-Z
40 MODE 0
50 FOR L=0 TO 15
60 INK L,L
70 NEXT L
80 BORDER 5:LOCATE 1,25:PAPER 5:PRINT"
  ";
90 PAPER INT(RND(1)*16)
100 PEN INT(RND(1)*16)
110 GOSUB 150
120 INK RND(1)*9+6,RND(1)*25
130 GOTO 90
150 REM PLOT KALEID POINTS
160 X=INT(RND(1)*20+1):Y=INT(RND(1)*24+1)
170 LOCATE X,Y:PRINT K$
180 LOCATE 21-X,Y:PRINT K$
190 LOCATE 21-X,25-Y:PRINT K$
200 LOCATE X,25-Y:PRINT K$
210 RETURN
```

ZOMBIE MAZE



Fingers at the ready!

Eyeballs glued to the screen!

Nerves stretched to the limit!

You are off and running through the maze as FOUR monsters converge on you for an inevitable end.

Stay alive as long as you can it's more fun that way.

How to play

Use the arrow keys to move your hunted little man around the maze for as long as you can. You have two moves to the monster's one – but there are four of them.

Programming hints

You might like to know that the tune being played behind your little man begins at line 790. If you are musical you could change it to the funeral march.

If you want to change the shape of the maze after a few games then alter the data in line 980 onwards. The 1 forms the walls and the 0 the gaps. Remember you must always have 20 characters on each line otherwise the maze will not be formed. You can have lots of gaps by using 0 but remember that this means the monsters can move faster as well as you.

The program

```

10 REM ZOMBIE MAZE
20 DEFINT A-Z
30 SC=0:HS=0
40 MODE 0
50 DEF FNSCR(X,Y)=TEST(32*X-20,408-16*Y)
60 ENV 1,1,120,7,24,-5,14
70 X=2:Y=2:FC=249
80 K$=""
90 MC=254
100 A(1)=19:B(1)=2:A(2)=2:B(2)=23
110 A(3)=9:B(3)=10:A(4)=19:B(4)=23
120 NUM=1
130 GOSUB 650
140 SOUND &81,0
150 SOUND &82,0
160 ON SQ(1) GOSUB 790
170 EVERY 10,0 GOSUB 410
180 EVERY 9,1 GOSUB 520
190 FLAG=0
200 REM MAIN LOOP
210 SC=SC+1:IF SC>HS THEN HS=SC
220 DI
230 LOCATE 1,25
240 PEN 8
250 PRINT "SCORE : ";
260 PRINT USING "#####";SC;
```

```

270 EI
280 IF FLAG=0 THEN GOTO 200
290 SOUND &C1,0
300 SOUND &C2,0
310 DI
320 Z=REMAIN(0):Z=REMAIN(1)
330 LOCATE 7,10:PEN 7:PRINT"GAME END"
340 FOR Z=1 TO 400:INK 7,INT(RND(1)*24):
NEXT Z:INK 7,13
350 FOR Z=1 TO 300:NEXT Z
360 IF SC=HS THEN GOSUB 860
370 A$=INKEY$:IF A$=" " THEN GOTO 370
380 A$=INKEY$:IF A$<>" " THEN GOTO 380
390 EI
400 SC=0:GOTO 40
410 REM PLAYER ROUTINE
420 IF INKEY(28)=0 THEN CY=-1
430 IF INKEY(63)=0 THEN CX=1
440 IF INKEY(30)=0 THEN CY=1
450 IF INKEY(71)=0 THEN CX=-1
460 DI
470 TX=X+CX:TY=Y+CY:CY=0:CY=0
480 COL=FNSCR(TX,TY):IF COL=5 THEN LOCAT
E X,Y:PRINT" ";X=TX:Y=TY:GOTO 490
490 PC=497-PC:LOCATE X,Y:PRINT CHR$(PC);

500 EI
510 RETURN
520 REM MONSTER ROUTINE
530 NUM=NUM+1:IF NUM=5 THEN NUM=1:MC=509
-MC
540 TRY=0
550 CA=SGN(X-A(NUM)):CB=SGN(Y-B(NUM))
560 TA=A(NUM)+CA:TB=B(NUM)+CB
570 COL=FNSCR(TA,TB):IF COL=5 THEN LOCAT
E A(NUM),B(NUM):PRINT" ";A(NUM)=TA:B(NU
M)=TB:GOTO 620
580 IF COL=3 THEN FLAG=-1:RETURN
590 TRY=TRY+1:IF TRY=3 THEN GOTO 620
600 CA=SGN(INT(RND*3)-1):CB=SGN(INT(RND*
3)-1)
610 GOTO 560
620 MC=509-MC:LOCATE A(NUM),B(NUM)
630 DI:PEN 11:PRINT CHR$(MC);:PEN 3
640 RETURN
650 REM DRAW THE MAZE

```


```

660 SYMBOL 240,&CC,&33,&CC,&33,&CC,&33,&
CC,&33
670 MODE 0:PAPER 5:PEN 12
680 BORDER 6
690 RESTORE 980
700 FOR L=1 TO 24
710 READ L$
720 FOR M=1 TO 20
730 D=VAL(MID$(L$,M,1))
740 IF D=1 THEN PRINT CHR$(240); ELSE PR
INT " ";
750 NEXT M
760 NEXT L
770 RESTORE
780 RETURN
790 REM PLAY NEXT NOTE OF TUNE
800 READ NOTE
810 IF NOTE=-1 THEN RESTORE:GOTO 800
820 SOUND &11,NOTE,50,0,1,0,0
830 SOUND &A,NOTE*0.69,50,0,1,0,0
840 ON SQ(1) GOSUB 790
850 RETURN
860 REM GOT HIGH SCORE
870 PEN 8
880 LOCATE 6,14:PRINT"HIGH SCORE"
890 FOR Z=1 TO 200 STEP 5:INK 8,INT(RND(
1)*24)
900 SOUND &83,ABS(500-Z*5),15
910 NEXT Z
920 RETURN
930 REM TUNE DATA
940 DATA 478,379,319,379
950 DATA 358,284,239,284
960 DATA 319,379,319,358,379,426,478,239
970 DATA -1
980 DATA"11111111111111111111"
990 DATA"10000001000010000001"
1000 DATA"10111100011000111101"
1010 DATA"10100101000010100101"
1020 DATA"10110001100110001101"
1030 DATA"10100100100100100101"
1040 DATA"10101101100110110101"
1050 DATA"10000101000010100001"
1060 DATA"11010001011010001011"
1070 DATA"10011111011011111001"
1080 DATA"10000010011001000001"

```

1090 DATA"10111000111100011101"
1100 DATA"10000010011001000001"
1110 DATA"10000000000000000001"
1120 DATA"10111010100101011101"
1130 DATA"10000010000001000001"
1140 DATA"11001110100101110011"
1150 DATA"10001010100101010001"
1160 DATA"10111011100111011101"
1170 DATA"10010001000010001001"
1180 DATA"11010101011010101011"
1190 DATA"10010101011010101001"
1200 DATA"10000100011000100001"
1210 DATA"11111111111111111111"

PONTOON



It's gamblers corner again as you pit your wits, and money, against the Amstrad croupier at the old classic game. By the way you can't cheat when you show your hand as your computer already knows what you are holding.

The object of the game, in case you have never strayed into a card school, is to achieve a score of 21 in as few cards as possible. If you over buy, or over twist, and score more than 21 you are 'bust' and you lose your stake. Aces count a 1 or 11 except when you receive two aces in a deal. In this case your hand will stand as 12.

How to play

The computer will deal your first two cards and ask if you would like to T (Twist, take another card without increasing your stake); S (Stay on your existing, or eventual hand); B (Buy another card); F (Fold if you think your hand is useless, but remember you will lose your stake money, and give you another hand).

If you buy, or twist, to a five card trick (five cards totalling 21 or less) you will beat the computer, unless it had a pontoon (21).

Press the key corresponding to your choice at each stage. There is no need to press ENTER at each stage.

When buying a card, type in the amount you wish to bet and press ENTER.

Remember, if you lose your shirt on this game you won't have to part with it.

The Program

```

10 REM PONTOON
20 MODE 2:BORDER 15
30 V=0
40 DIM C(52)
50 DIM P(10),M(2)
60 GOSUB 760
70 M=100
80 REM HAND LOOP
90 CLS
100 M=M-2:IF M<0 THEN GOTO 700
110 PRINT"You have ";M;" pounds left"
120 BET=2
130 PRINT"Your cards : "
140 GOSUB 900
150 GOSUB 980:PRINT" The "K$
160 P(1)=V:GOSUB 980:GOSUB 900
170 PRINT " The ";K$:P(2)=V:GOSUB 900
180 N=2
190 IF P(1)=1 THEN P(1)=11
200 IF P(2)=1 AND P(1)<>11 THEN P(2)=11
210 IF P(1)+P(2)=21 THEN PRINT:PRINT" Yo
u have got a PONTOON !!!"
220 SP=P(1)+P(2)
230 REM PLAYER CHOICES
240 PRINT "Twist / Stick / Buy / Fold  ?
"
250 A$=INKEY$:IF A$<>"" THEN GOTO 250
260 A$=INKEY$:IF A$="" THEN GOTO 260

```



```

270 IF A$<>"T" AND A$<>"S" AND A$<>"B" A
ND A$<>"F" THEN GOTO 250
280 IF A$="F" THEN PRINT"O.K.  You lose
your stake":GOSUB 900:GOTO 90
290 IF A$="S" THEN GOTO 400
300 IF A$="B" THEN GOSUB 930:PRINT"O.K.
  Good luck"
310 N=N+1:REM NEXT CARD
320 GOSUB 900:GOSUB 980
330 PRINT"The ";K$
340 P(N)=V:SP=0
350 FOR L=1 TO N:SP=SP+P(L):NEXT L
360 IF SP>21 THEN PRINT"BUST.....":GOT
O 400
370 IF N>4 THEN PRINT"FIVE CARD TRICK...
... "
380 IF SP=21 THEN PRINT"TWENTY ONE.....
. "
390 GOTO 210
400 REM Now the computer plays
410 SP=0
420 FOR L=1 TO N:SP=SP+P(L):NEXT L
430 PRINT:PRINT"My cards : "
440 GOSUB 900
450 MT=0:GOSUB 980:PRINT" The ";K$
460 M(1)=V:GOSUB 900:GOSUB 980
470 PRINT" The ";K$:GOSUB 900:M(2)=V
480 IF M(1)=1 THEN M(1)=11
490 IF M(2)=1 AND M(1)<>11 THEN M(2)=11
500 IF M(1)+M(2)=21 THEN GOTO 610
510 IF P(1)+P(2)=21 AND N=2 THEN GOTO 65
0
520 MT=M(1)+M(2)
530 IF MT<SP AND SP<=21 THEN GOTO 550
540 IF N<5 THEN PRINT"I will stick ":GOS
UB 900:GOTO 610
550 PRINT"I will twist..."
560 GOSUB 980:GOSUB 900
570 PRINT" The ";K$
580 MT=MT+V:GOSUB 900
590 IF MT>21 THEN PRINT"I am bust.....":
GOTO 650
600 GOTO 530
610 REM PLAYER LOSES
620 PRINT:PRINT"You lose that hand"
630 GOSUB 900:GOSUB 900

```


```

640 GOTO 80
650 REM COMPUTER LOSES
660 PRINT:PRINT"You win that hand"
670 GOSUB 900:GOSUB 900
680 M=M+2*BET
690 GOTO 80
700 REM HERE WHEN PLAYER IS BROKE
710 PRINT:PRINT"You are broke"
720 GOSUB 900
730 PRINT"Come back on a luckier day"
740 GOSUB 900
750 END
760 REM RANDOMIZE PACK
770 PRINT"+++++++ NEW PACK ++++++"
780 FOR L=1 TO 52
790 C(L)=L:NEXT L
800 REM SHUFFLE
810 FOR L=1 TO 100
820 IF V>10 THEN V=10
830 A=INT(RND(1)*52+1)
840 B=INT(RND(1)*52+1)
850 T=C(A):C(A)=C(B)
860 C(B)=T
870 NEXT L
880 Q=1
890 RETURN
900 REM A SHORT DELAY
910 FOR ZZ=1 TO 700:NEXT ZZ
920 RETURN
930 REM INCREASE THE BET BY BUYING
940 PRINT"BUY for how much ";
950 INPUT BB:IF M-BB<0 THEN PRINT"You ha
ven't got that much money":GOTO 940
960 BET=BET+BB:M=M-BB
970 RETURN
980 REM CARD IN K
990 REM NAME IN K$
1000 K=C(Q):Q=Q+1:IF Q=53 THEN GOSUB 760
1010 S=INT((K-1)/13):V=K-13*S
1020 ON V GOSUB 1060,1070,1080,1090,1100
,1110,1120,1130,1140,1150,1160,1170,1180
1030 ON S+1 GOSUB 1190,1200,1210,1220
1040 IF V>10 THEN V=10
1050 RETURN
1060 K$="Ace ":RETURN
1070 K$="Two ":RETURN
1080 K$="Three ":RETURN

```

```
1090 k$="Four ":RETURN
1100 k$="Five ":RETURN
1110 k$="Six ":RETURN
1120 k$="Seven ":RETURN
1130 k$="Eight ":RETURN
1140 k$="Nine ":RETURN
1150 k$="Ten ":RETURN
1160 k$="Jack ":RETURN
1170 k$="Queen ":RETURN
1180 k$="King ":RETURN
1190 k$=k$+"of Clubs":RETURN
1200 k$=k$+"of Diamonds":RETURN
1210 k$=k$+"of Hearts":RETURN
1220 k$=k$+"of Spades":RETURN
```

MORSE ENCODER



If you are planning to go sailing in the near future, then this program will help you brush up your morse code before you go.

Can you remember what to tap for S.O.S.?

It's . . . - - - . . .

That's all the help we are going to give you. Hope you don't have to use it.

When you type a message into this program, your computer will play it back in morse code. A good way to test each message is to have someone else enter the message, and for you to sit down on the other side of the room and see if you can decode it.

How to play

Just press the alphabet keys of your choice.

Programming Hint

If you feel that the play back is too slow, or too fast, you can change the speed by altering $S=12$ in line 20 of the program.

The program


```

10 REM MORSE ENCODER
20 S=12:D=4
30 DIM M$(26)
40 FOR L=1 TO 26:READ M$(L)
50 NEXT L
60 MODE 1
70 BORDER 13
80 LOCATE 1,6
90 PRINT "Message to transmit : "
100 LINE INPUT "",I$
110 FOR L=1 TO LEN(I$)
120 CH$=MID$(I$,L,1)
130 IF CH$>="a" AND CH$<="z" THEN CH$=CHR$(ASC(CH$)-ASC("a")+ASC("A"))
140 IF CH$<"A" OR CH$>"Z" THEN GOSUB 210
:GOTO 160
150 GOSUB 350
160 NEXT L
170 LOCATE 1,20:PRINT"Transmission complete"
180 FOR L=1 TO 2000:NEXT L
190 CALL &BB06
200 RUN
210 REM PAUSE BETWEEN WORDS
220 FOR W=1 TO 120*D:NEXT W
230 RETURN
240 REM SHORT PAUSE
250 FOR W=1 TO 30*D:NEXT W
260 RETURN
270 REM DOT

```

```
280 SOUND 1,130,3*D:SOUND 1,0,D
290 GOSUB 240
300 RETURN
310 REM DASH
320 SOUND 1,130,9*D:SOUND 1,0,D
330 GOSUB 240
340 RETURN
350 REM SOUND ONE CHAR
360 X=ASC(CH$)-ASC("A")+1
370 F$=M$(X)
380 FOR M=1 TO LEN(F$)
390 IF MID$(F$,M,1)="S" THEN GOSUB 270:G
OTO 410
400 GOSUB 310
410 NEXT M
420 RETURN
430 DATA SL,LSSS,LLS,LSS,S,SSLs,LLS
440 DATA SSSS,SS,SLLL,LSL,SLSS,LL
450 DATA LS,LLL,SLLS,LLSL,SLS,SSS,L
460 DATA SSL,SSSL,SLL,LSSL,LSLL,LLSS
```

SAFECRACKER



Here is your chance to see if you could be of any use to Smiley's People as an inside man.

We have set up a series of numeric codes for you to 'crack' in order to open the safe.

Digits range from 1 to 8 inclusive.

How to play

Your computer will select, at random, it's own sequence and ask for your first choice of four numbers.

Enter your numbers and press ENTER.

After your first 'guess' your Amstrad CPC will put up it's 'mark' to show you how many of your numbers are correct, and how many are in the right place.

You will not be told **which** of your numbers is in the correct place, only that one, or more, is in the correct place. You will still have to decide for your next move which of your first selection is correct.

We told you it was a test for Smiley's People didn't we.

The Program

```

10 REM SAFECRACKER
20 DIM C(4),G(4),T(4)
30 GOSUB 270
40 MODE 1:BORDER 8
50 G=0:F=0
60 PEN 1
70 GOSUB 360
80 G=G+1:IF G=11 THEN GOTO 190
90 GOSUB 390
100 GOSUB 550
110 IF F=0 THEN GOTO 80
120 REM CORRECT DEDUCTION
130 LOCATE 1,19:PRINT"CORRECT ... The Sa
fe swings gently open"
140 FOR L=1 TO 4
150 PRINT C(L);" ";
160 NEXT L
170 LOCATE 9,21:PRINT "IN ";G;" attempts
"
180 GOTO 240
190 REM THE COMBINATION REMAINED HIDDEN
200 LOCATE 1,14:PRINT"Alarm bells ring o
ut loudly":LOCATE 1,16:PRINT"The time sw
itch has been triggered"
210 LOCATE 1,19:PRINT"The combination wa
s : "
220 LOCATE 1,21:PRINT"          ";
230 FOR L=1 TO 4:PRINT C(L);" ";:NEXT L
240 IF INKEY$=" " THEN GOTO 240
250 IF INKEY$<>" " THEN GOTO 250
260 RUN
270 REM SET THE COMBINATION LOCK
280 C(1)=INT(RND(1)*8+1)

```




```

290 FOR L=2 TO 4
300 C(L)=INT(RND(1)*8+1)
310 FOR I=1 TO L-1
320 IF C(L)=C(I) THEN C(L)=INT(RND(1)*8+
1)
330 NEXT I
340 NEXT L
350 RETURN
360 REM TITLE THE COLUMNS
370 LOCATE 4,1:PRINT"YOUR GUESS   my mark
"
380 RETURN
390 REM GET A GUESS INTO G()
400 L=1
410 IF INKEY$<>" " THEN GOTO 410
420 D$=INKEY$:IF D$="" THEN GOTO 420
430 D=ASC(D$)
440 IF D=127 THEN GOTO 500
450 IF D=13 THEN GOTO 530
460 IF D<ASC("1") OR D>ASC("9") THEN GOT
O 410
470 IF L=5 THEN GOTO 410
480 LOCATE 3+L*2,G+3:PEN 2:PRINT CHR$(D)
;
490 G(L)=D-48:L=L+1:GOTO 410
500 L=L-1:IF L=0 THEN L=1
510 LOCATE 3+L*2,G+3:PRINT " ";
520 GOTO 410
530 IF L<5 THEN GOTO 410
540 RETURN
550 REM SCORE G() AGAINST THE COMBINATIO
N
560 B1=0:W1=0
570 FOR H=1 TO 4
580 IF G(H)=C(H) THEN B1=B1+1
590 NEXT H
600 FOR L=1 TO 4:T(L)=C(L):NEXT L
610 FOR J=1 TO 4
620 IF T(J)=-1 THEN GOTO 660
630 FOR L=1 TO 4
640 IF T(J)=G(L) THEN W1=W1+1:T(J)=-1
650 NEXT L
660 NEXT J
670 W1=W1-B1
680 IF B1=4 THEN F=1

```

```
690 LOCATE 15,3+6  
700 PEN 3  
710 PRINT B1;" ";W1  
720 RETURN
```

BIORYTHM GRAPHING AND PREDICTION



Don't worry it isn't going to hurt you, just lie back and enjoy the sensation of your life flashing past you.

In simple terms this is a system which will allow you to predict the type of day you are going to have, or check and see if the day just passed, is as bad as you think possible.

You type in the data and your Amstrad CPC will tell you, and show you the shape of the day to come. It will also give you an idea of the days on both sides of the day in question.

Imagine the potential in drawing up friend's charts and sending them as Christmas or birthday presents.

How to play

The Amstrad CPC will ask you to enter your birth date and the date of the day on which you are using the program.

You will then be asked for the year, month and day/date of the week you wish to predict. When you are asked for the day remember to enter the day/date not just any day as there are, of course, several Tuesdays in a month.

Press ENTER key after each entry.

After you key in the final information the computer will produce a chart of your physical, emotional and intellectual state on the day in question. As we said before, you will also see a picture of the days on either side and you may then decide not to leave your bed for the next week.

To clear, type RUN and return to the menu.

The Program

```

10 REM BIORHYTHM GRAPHING AND PREDICTION
20 MODE 1:BORDER 2
30 LOCATE 1,5:PRINT"BIRTHDATE ENTRY : "
40 GOSUB 290
50 BY=Y:BM=M:BD=D
60 CLS:LOCATE 1,5:PRINT"DATE TO INVESTIG
ATE ENTRY : "
70 GOSUB 290
80 GOSUB 430
90 EM=DAYS-28*INT(DAYS/28)
100 IT=DAYS-33*INT(DAYS/33)
110 FH=DAYS-23*INT(DAYS/23)
120 CLS:PEN 1:LOCATE 6,1
130 PRINT"PHYSICAL CYCLE"
140 O=PH/23*360:MU=28/23
150 C=1:GOSUB 550
160 LOCATE 6,2:PEN 2:C=2
170 PRINT"EMOTIONAL CYCLE"
180 O=EM/28*360:MU=1
190 GOSUB 550
200 LOCATE 6,3:PEN 3:C=3
210 PRINT"INTELLECTUAL CYCLE"
220 O=IT/33*360:MU=28/33
230 GOSUB 550
240 PLOT 384,10:DRAW 384,250

```

```

250 FEN 1
260 IF INKEY$<>" " THEN GOTO 260
270 IF INKEY$="" THEN GOTO 270
280 RUN
290 REM READ IN A DATE
300 LOCATE 1,11:PRINT"YEAR.....
      ":LOCATE 10,11
310 INPUT Y:IF Y<>INT(Y) THEN GOTO 300
320 LOCATE 1,16:PRINT"MONTH....
      ":LOCATE 10,16

330 INPUT M$
340 IF LEFT$(M$,1)>"0" AND LEFT$(M$,1)<=
"9" THEN M=VAL(M$):GOTO 390
350 RESTORE 640
360 M=0:FOR L=1 TO 12
370 READ T$:IF T$=LEFT$(M$,3) THEN M=L
380 NEXT L
390 IF M=0 OR M<>INT(M) OR M>12 THEN GOT
0 320
400 LOCATE 1,21:PRINT"DAY.....
      ":LOCATE 10,21
410 INPUT D:IF D<1 OR D>31 OR D<>INT(D)
THEN GOTO 400
420 RETURN
430 REM FIND THE NUMBER OF DAYS
440 DP=0:DM=0
450 DP=365.25*Y:DM=365.25*BY
460 RESTORE 680
470 FOR L=1 TO 12
480 READ T
490 IF M=L THEN DP=DP+T
500 IF BM=L THEN DM=DM+T
510 NEXT L
520 DP=DP+D:DM=DM+BD
530 DAYS=DP-DM
540 RETURN
550 REM DRAW A CYCLE
560 Y=70+60*SIN((-360*MU+0)*PI/180)
570 X=48
580 FOR TH=-360 TO 350 STEP 2.5
590 OX=X:X=TH/4.5+128:OY=Y
600 Y=70+60*SIN((TH*MU+0)*PI/180)
610 PLOT 3*OX,2*OY,C:DRAW 3*X,2*Y:NEXT T
H
620 FOR L=1 TO 3000:NEXT L
630 RETURN

```

640 DATA JAN, FEB, MAR


650 DATA APR, MAY, JUN

660 DATA JUL, AUG, SEP

670 DATA OCT, NOV, DEC

680 DATA 0, 31, 59, 90, 120, 151, 181, 212, 243,
273, 304, 334

DICE



If you are bored with shooting down missiles and defending the world from alien attack then here is a game to test your gambling skill.

The Amstrad CPC will 'throw' five dice for you and you must try to achieve the best sequence possible for the greatest score.

Possibilities to achieve highest scores are:

- One pair
- Three of a kind
- Two pair
- Full house (Three of a kind and two of a kind)
- Four of a kind
- Low run
- High run
- Five of a kind

How to play

After the computer's first 'throw' you can hold the dice by using keys 1 to 5. Press R to release all the dice. Use SPACE to roll the dice free. You have ten rolls to make your high score and give your partner a target to beat.

The Program

```

10 REM DICE
20 MODE 1:BORDER 12
30 DIM D(5),F(5),S(6)
40 T=10
50 SC=0
60 GOSUB 1110
70 PEN 1
80 T=T-1:IF T=0 THEN GOTO 490
90 FOR L=1 TO 5:F(L)=0:NEXT L
100 LOCATE 1,7:PRINT"
      "
110 LOCATE 1,11:PRINT"
      "
120 LOCATE 1,15:PRINT"
      "
130 LOCATE 1,20:PRINT"
      "
140 GOSUB 220
150 GOSUB 350
160 GOSUB 220
170 GOSUB 350
180 GOSUB 220
190 GOSUB 620
200 SC=SC+NS
210 GOTO 80
220 REM GENERATE AND SHOW FIVE DICE
230 FOR L=1 TO 5
240 IF F(L)=1 THEN GOTO 330
250 Q=INT(RND(1)*60+10)
260 D(L)=1
270 FOR E=1 TO Q
280 LOCATE L*3+4,5
290 D(L)=D(L)+1

```



```

300 IF D(L)=7 THEN D(L)=1
310 PRINT CHR$(128+D(L));
320 NEXT E
330 NEXT L
340 RETURN
350 REM SET THE STOP FLAGS
360 LOCATE 8,2:PRINT"HOLD NOW !!!"
370 LOCATE 1,7:PRINT"
      "
380 FOR L=1 TO 5:F(L)=0:NEXT L
390 A$=INKEY$:IF A$<>" " THEN GOTO 390
400 A$=INKEY$:IF A$="" THEN GOTO 400
410 IF A$=" " THEN GOTO 460
420 IF A$="R" THEN GOTO 370
430 IF A$<"1" OR A$>"5" THEN GOTO 390
440 A=ASC(A$)-48:LOCATE 3*A+4,7:PRINT"X"
450 F(A)=1:GOTO 400
460 REM FINISHED
470 LOCATE 8,2:PRINT"
      "
480 RETURN
490 REM END OF GAME
500 MODE 1:PEN 3
510 LOCATE 1,11
520 PRINT"End of the Game"
530 LOCATE 1,14
540 PRINT"Your score is : "
550 LOCATE 9,17
560 PRINT SC
570 LOCATE 6,23
580 PRINT"Press <SPACE> to continue"
590 a$=INKEY$:IF A$=" " THEN GOTO 590
600 A$=INKEY$:IF A$<>" " THEN GOTO 600
610 RUN
620 REM SCORE THE DICE
630 REM RETURN SCORE IN NS
640 FOR L=1 TO 6:S(L)=0:NEXT L
650 FOR L=1 TO 5
660 S(D(L))=S(D(L))+1
670 NEXT L
680 Z2=0:FOR L=1 TO 6
690 IF S(L)=1 THEN Z2=Z2+1
700 NEXT L
710 Z1=0:FOR L=1 TO 6
720 IF S(L)=0 THEN Z1=Z1+1
730 NEXT L

```

```

740 IF Z1=5 THEN GOSUB 930:GOTO 840
750 IF Z1=1 AND S(6)=0 THEN GOSUB 1050:G
OTO 840
760 IF Z1=1 AND S(1)=0 THEN GOSUB 1070:G
OTO 840
770 IF Z1=1 THEN GOSUB 1090:GOTO 840
780 IF Z1=2 THEN GOSUB 1030:GOTO 840
790 IF Z1<>3 THEN GOTO 820
800 IF Z2=1 THEN GOSUB 970:GOTO 840
810 IF Z2=2 THEN GOSUB 1010:GOTO 840
820 IF Z2=1 THEN GOSUB 990:GOTO 840
830 GOSUB 950
840 REM SCORE IS NOW IN NS
850 REM AND MESSAGE IS IN N$
860 LOCATE 1,11:PRINT"
"
870 LOCATE 1,11:PRINT N$
880 LOCATE 9,15:PRINT"SCORE :";NS;"
"
890 LOCATE 16,20:PRINT"<SPACE>"
900 IF INKEY$=" " THEN GOTO 900
910 IF INKEY$<>" " THEN GOTO 910
920 RETURN
930 REM FIVE OF A KIND
940 N$="***** EUREKA *****":NS
=50:RETURN
950 REM PAIR AND THREE
960 N$="***** FULL HOUSE *****":N
S=30:RETURN
970 REM TWO PAIRS
980 N$="***** DOUBLE DOUBLE *****":NS=
20:RETURN
990 REM FOUR OF A KIND
1000 N$="***** FOUR TIMES *****":N
S=15:RETURN
1010 REM THREE OF A KIND
1020 N$="***** THREE TIMES *****":NS
=10:RETURN
1030 REM ONE PAIR
1040 N$="***** MADE FOR EACH OTHER *****":
NS=7:RETURN
1050 REM LOW RUN
1060 N$="***** LOW RUN *****":NS
=24:RETURN
1070 REM HIGH RUN
1080 N$="***** HIGH RUN *****":N

```

```
S=40:RETURN
1090 REM NO SCORE AT ALL
1100 Z$="***** ZERO *****":N
S=0:RETURN
1110 REM DEFINE DICE SHAPES
1120 SYMBOL AFTER 129
1130 SYMBOL 129,0,0,0,24,24,0,0,0
1140 SYMBOL 130,3,3,0,0,0,0,192,192
1150 SYMBOL 131,3,3,0,24,24,0,192,192
1160 SYMBOL 132,195,195,0,0,0,0,195,195
1170 SYMBOL 133,195,195,0,24,24,0,195,195
1180 SYMBOL 134,195,195,0,195,195,0,195,195
1190 RETURN
```

HUNT THE DAY



Have you ever wondered if your old dad was 'Fair of Face', 'Full of Grace', 'Full of Woe', had 'Far to Go' or, hopefully, was 'Full of Loving and Giving'?

Now is your chance to find out.

If you ask any members of your family, or your friends, when they were born you will be able to tell them, using this program, the actual day of the week when the great event happened.

By the way, this chart is only effective as far back as 1st January, 1753 which was when the calendar changed to it's present form. Of course, you already knew that didn't you?

Oh, by the way, we have taken account of leap years just in case you thought you could catch us out.

How to play

You will be asked for details of the date you have in mind.

When you have typed in the data press ENTER after each item.

When you have completed the requested material your Amstrad CPC will do the rest for you.

The Program

```

10 REM HUNT THE DAY
100 MODE 1
110 LOCATE 1,4:PRINT"Please give the dat
e you wish to ":PRINT"investigate."
120 GOSUB 1000
130 GOSUB 2000
140 LOCATE 1,2
150 PRINT"      That was/is/will be a ";T
$
160 LOCATE 1,21:PRINT"Please press <SPAC
E> when ready"
200 IF INKEY$="" " THEN GOTO 200
210 IF INKEY$<>" " THEN GOTO 210
299 RUN
1000 REM get a date
1010 LOCATE 1,11:PRINT"YEAR....
      ":LOCATE 10,11
1020 INPUT y:IF y<1753 OR y<>INT(y) THEN
GOTO 1010
1100 LOCATE 1,14:PRINT"MONTH...
      ":LOCATE 10,14
1110 INPUT M$:IF LEFT$(M$,1)>="1" AND LE
FT$(M$,1)<="9" THEN M=VAL(M$):GOTO 1200
1120 RESTORE 9000
1125 M=0
1130 FOR L=1 TO 12
1140 READ T$
1150 IF T$=LEFT$(M$,3) THEN M=L
1160 NEXT L
1200 IF M<1 OR M>12 OR M<>INT(M) THEN GO
TO 1100

```

```

1300 LOCATE 1,17:PRINT"DAY.....
                                ":LOCATE 10,17
1310 INPUT D
1320 IF D<1 OR D>31 OR D<>INT(D) THEN GO
TO 1300
1999 RETURN
2000 REM FIND OUT DAY OF WEEK
2010 A=Y+1+31*(M-1)+INT((Y-1)/4)-INT(3*I
NT((Y+99)/100)/4)
2020 IF M<=2 THEN GOTO 2040
2030 A=Y+1+31*(M-1)-INT((4*M+23)/10)+INT
(Y/4)-INT(3*(INT(Y/100)+1)/4)
2040 A=A-INT(A/7)*7
2050 IF A=0 THEN A=7
2060 A=A-1:IF M=2 THEN GOTO 2100
2070 N=31
2080 IF M=4 OR M=6 OR M=9 OR M=11 THEN N
=30
2090 GOTO 2120
2100 N=28
2110 IF Y-INT(Y/4)*4=0 THEN N=29
2120 IF D>N THEN T$="NON-DAY
.....DOESN'T EXIST.....":RETURN
2130 A=A+D:D=A-INT(A/7)*7:IF D=0 THEN D=
7
2135 RESTORE 9500
2140 FOR L=1 TO D
2150 READ T$
2160 NEXT L
2999 RETURN
9000 DATA JAN,FEB,MAR
9010 DATA APR,MAY,JUN
9020 DATA JUL,AUG,SEP
9030 DATA OCT,NOV,DEC
9500 DATA SUNDAY,MONDAY,TUESDAY
9510 DATA WEDNESDAY,THURSDAY,FRIDAY
9520 DATA SATURDAY

```

WALLBREAKER



This is our version of the popular arcade game of the same name. Along the top of the screen you will see three rows of coloured 'bricks' and you must use the bouncing ball to demolish the wall in as few moves as possible.

A little hint. If you break through the first two rows and get to the back wall the ball will stay bouncing, and demolishing, from behind until it breaks out again.

How to play

Use arrow keys to move the bat across the screen.

Programming Hints

If you are finding the game too easy, or too difficult, you can change the size of the bat by altering Line 80. Increase, or reduce the number after BL=. If you want to sit back and let

the ball bounce on it's own then simply increase the bat size until it fills the bottom of the screen.

The Program

```

10 REM
20 ENV 1,1,15,1,15,-1,1
30 GOSUB 590
40 HS=0
50 BL=7
60 BA$="":FOR L=1 TO BL:BA$=BA$+CHR$(132)
70 BE=31-BL
80 GOSUB 450
90 REM MAIN NEW GAME LOOP
100 SC=0
110 REM MAIN NEW FRAME LOOP
120 BH=90
130 GOSUB 360
140 BX=10:GOSUB 540
150 X=5:Y=20
160 CX=1:CY=-1
170 GF=1
180 REM MAIN BOUNCE AND SCORE LOOP
190 GOSUB 750
200 IF GF=0 THEN GOTO 1000
210 GOSUB 920
220 GOSUB 960
230 PEN 3
240 LOCATE 8,1:PRINT SC;
250 IF BH<>0 THEN GOTO 190
260 REM END OF FRAME
270 FOR L=10 TO 250 STEP 5
280 SOUND 1,L,5:SOUND 2,L+7,4
290 NEXT L
300 MODE 1:LOCATE 4,11
310 PRINT "Your score is now :";SC
320 IF SC>HS THEN PRINT"You have the high score":PRINT:PRINT"and you're still playing":GOTO 340
330 PRINT"You are ";hs-sc;" points behind the high score"
340 FOR L=1 TO 4000:NEXT L

```



```

350 GOTO 110
360 REM Draw screen layout
370 MODE 1:BORDER 14:PAPER 0:PEN 2
380 FOR L=2 TO 22
390 LOCATE 1,L:PRINT CHR$(129):LOCATE 32
,L:PRINT CHR$(130);:NEXT L
400 LOCATE 2,2:FOR L=2 TO 31:PRINT CHR$(
134);:NEXT L
410 FOR L=2 TO 31:PEN 3:LOCATE L,6:PRINT
CHR$(133);:LOCATE L,13:PRINT CHR$(133);
420 PEN 1:LOCATE L,9:PRINT CHR$(133);:NE
XT L
430 LOCATE 2,1:PRINT"SCORE:          HIGH
: ";HS
440 RETURN
450 REM GENERATE THE PATTERNS
460 SYMBOL AFTER 129
470 SYMBOL 129,4,2,1,2,4,2,1,2
480 SYMBOL 130,32,64,128,64,32,64,128,64
490 SYMBOL 131,24,62,126,255,255,126,62,
24
500 SYMBOL 132,255,255,170,85,72,16,0,0
510 SYMBOL 133,255,170,85,170,85,170,85,
255
520 SYMBOL 134,0,0,0,0,0,0,0,255
530 RETURN
540 REM DRAW BAT ON SCREEN
550 LOCATE BX,23
560 PEN 3
570 PRINT BA$;
580 RETURN
590 MC=40000
600 TA=40000+&7F
610 TH=INT(TA/256)
620 TL=TA-256*TH
630 POKE MC,&CD
640 POKE MC+1,&60
650 POKE MC+2,&BB
660 POKE MC+3,&32
670 POKE MC+4,TL
680 POKE MC+5,TH
690 POKE MC+6,&C9
700 RETURN
710 LOCATE XX,YY
720 CALL MC
730 DD=PEEK(TA)

```

```


740 RETURN
750 REM SET UP NEW BALL POSITION
760 REM CHANGE SCORE IF APPROPRIATE
770 C=X+CX:D=Y+CY
780 IF C<2 OR C>30 THEN CX=-CX:SC=SC+2:G
OTO 770
790 IF D<3 THEN CY=-CY:SC=SC+4:CX=INT(RN
D(1)*3)-1:GOTO 770
800 IF D>23 THEN GF=0:RETURN
810 XX=C+1:YY=D+1:GOSUB 710
820 IF DD=32 THEN RETURN
830 SOUND 1,280,20,0,1
840 IF DD=132 THEN GOTO 880
850 BH=BH-1
860 SC=SC+D:LOCATE XX,YY:PRINT " ";
870 CX=INT(RND(1)*3)-1:CY=-CY:GOTO 770
880 SC=SC+20:CY=-CY
890 CX=INT(RND(1)*3)-1
900 GOTO 770
910 RETURN
920 REM ERASE OLD BALL, PLOT NEW BALL
930 LOCATE C+1,D+1:PRINT CHR$(131);:LOCA
TE X+1,Y+1:PRINT " "
940 X=C:Y=D
950 RETURN
960 REM MOVE BAT IF NECESSARY
970 IF INKEY(8)=0 AND BX>1 THEN BX=BX-1:
GOSUB 540
980 IF INKEY(1)=0 AND BX<BE THEN BX=BX+1
:GOSUB 540
990 RETURN
1000 REM PLAYER HAS LOST THE BALL
1010 CLS:LOCATE 1,10:PRINT"GAME OVER"
1020 LOCATE 1,15:PEN 2:PRINT"YOU SCORED
:";SC
1030 IF SC>HS THEN LOCATE 1,18:PRINT"WHI
CH IS THE HIGH SCORE":HS=SC:GOTO 1050
1040 LOCATE 1,18:PRINT"SLIGHTLY (";HS-SC
;") less than ":LOCATE 1,20:PRINT"the hi
gh score"
1050 LOCATE 1,21:PRINT"DO YOU WANT TO FL
AY AGAIN ":LOCATE 15,23:PRINT"(Y/N)"
1060 I$=INKEY$
1070 IF I$<>"Y" AND I$<>"N" THEN GOTO 10
60
1080 IF I$="Y" THEN GOTO 100

```

130

```
1090 MODE 1:LOCATE 11,11:PRINT"type RUN  
to restart"  
1100 END
```

TECHNICAL ARTIST



If you have dreams of becoming an architect and are interested in producing your own plans for rooms, gardens or garages then this is the program for you.

Using this program you will be able to produce highly accurate line drawings which you can then 'dump' onto a printer, if you have one and hand on to a planner. You could use this item for estimating how much carpeting you are going to need for new rooms etc.

You can also change the colours to give you an idea of the final effect.

How to play

To move the cursor around the screen use the ARROW keys.

To fix the end point of a line press key F.

To release a line and allow the cursor to move freely press key R.

To clear the screen press Z.

To change the colour with which you are drawing press C. You will then have to choose another colour by using digit 0 to 3 and by pressing ENTER.

The program


```

10 REM TECHNICAL ARTIST
20 MODE 1
30 GOSUB 390
40 PRINT CHR$(817);CHR$(1);
50 X=320:Y=200
60 FX=0
70 WINDOW #1,1,40,1,24
80 WINDOW #2,1,40,25,26
90 COLOUR=1
100 PLOT X,Y,COLOUR
110 REM MAIN COMMAND LOOP
120 CX=0:CY=0
130 IF INKEY(0)=0 THEN CY=4
140 IF INKEY(1)=0 THEN CX=4
150 IF INKEY(2)=0 THEN CY=-4
160 IF INKEY(8)=0 THEN CX=-4
170 TX=X+CX:TY=Y+CY
180 IF TX<1 OR TX>639 THEN TX=X
190 IF TY<17 OR TY>399 THEN TY=Y
200 A$=INKEY$:IF A$="" THEN GOTO 260
210 IF INKEY(53)=0 THEN FX=1:XFIX=X:YFIX=Y
220 IF INKEY(71)=0 THEN CLS #1:PLOT X,Y,COLOUR
230 IF INKEY(62)=0 THEN GOSUB 450
240 IF INKEY(50)=0 THEN FX=0:PLOT X,Y,COLOUR
250 GOTO 110
260 REM MOVE CURSOR
270 IF FX=1 THEN GOTO 320
280 PLOT X,Y,COLOUR

```

```
290 PLOT TX, TY
300 X=TX:Y=TY
310 GOTO 110
320 REM MOVE RUBBER BAND
330 IF X=TX AND Y=TY THEN GOTO 110
340 MOVE X, Y
350 DRAW XFIX, YFIX, COLOUR
360 DRAW TX, TY
370 X=TX:Y=TY
380 GOTO 110
390 REM START-UP KEY DEFINITIONS
400 KEY DEF 0,0
410 KEY DEF 1,0
420 KEY DEF 2,0
430 KEY DEF 8,0
440 RETURN
450 REM CHANGE THE COLOUR SUBROUTINE
460 CLS #2
470 LOCATE 1,25
480 PRINT"COLOUR : 0"
490 COLOUR$="0"
500 A$=INKEY$
510 IF A$=CHR$(13) THEN GOTO 560
520 IF A$<"0" OR A$>"9" THEN GOTO 500
530 COLOUR$=A$
540 LOCATE 9,25:PRINT VAL(COLOUR$); " ";
550 GOTO 500
560 COLOUR=VAL(COLOUR$)
570 IF COLOUR>3 THEN GOTO 460
580 RETURN
```

SPIDER



Now here is a variation on the game of 'Frogger' which you have probably seen on expensive tapes and in arcade games.

You have to help a spider back to it's web in the top of the screen by moving it through waves of cars. Every time you manage to get the spider home another wave begins so keep your wits about you.

How to play

Use the ARROW keys to move the spider around and when your little friend is killed off restart by pressing the SPACE bar.

The program

```

10 REM SPIDER
20 BORDER 13
30 GOSUB 620
40 GOSUB 780
50 REM NEW PLAYER START
60 SCORE=0
70 REM NEW WAVE START
80 MODE 0
90 WAVE=WAVE+1
100 TM=300
110 GOSUB 840
120 GOSUB 940
130 X=15:Y=24
140 PEN 11
150 LOCATE X,Y:PRINT CHR$(251);
160 GOSUB 1080
170 TM=TM-6
180 LOCATE X,Y:GOSUB 740
190 IF DD<>251 AND DD<>206 OR TM=0 THEN
GOTO 370
200 XX=X:YY=Y
210 IF INKEY(0)=0 THEN YY=Y-1
220 IF INKEY(1)=0 THEN XX=X+1
230 IF INKEY(2)=0 THEN YY=Y+1
240 IF INKEY(8)=0 THEN XX=X-1
250 IF X<1 OR X>20 OR Y>25 THEN GOTO 370
260 LOCATE XX,YY:GOSUB 740
270 IF DD=253 THEN GOTO 500
280 IF DD<>206 AND DD<>251 THEN GOTO 370
290 LOCATE X,Y
300 PEN 7
310 PRINT CHR$(206);
320 LOCATE XX,YY
330 PEN 11
340 PRINT CHR$(251);
350 X=XX:Y=YY
360 GOTO 160
370 REM DEAD
380 LOCATE 4,25:PEN 14
390 PRINT" SPLAT ! ";
400 LOCATE 3,10
410 PRINT" SCORE : ";SCORE

```



```

420 FOR Q=300 TO 450 STEP 50
430 SOUND 1,Q,50
440 SOUND 2,Q+15,45
450 SOUND 0,Q+30,40
460 NEXT Q
470 IF INKEY$=" " THEN GOTO 470
480 IF INKEY$("<>") THEN GOTO 480
490 GOTO 60
500 REM SPIDER HAS REACHED THE WEB
510 SCORE=SCORE+TM
520 PEN 7
530 LOCATE 10,2
540 PRINT CHR$(206);
550 PEN 15
560 LOCATE 10,1:PRINT CHR$(251);
570 LOCATE 5,25:PRINT"WELL DONE"
580 LOCATE 3,10:PRINT" SCORE : ";SCORE
590 FOR Q=1 TO 500
600 SOUND 1,Q,1:NEXT Q
610 GOTO 70
620 REM MACHINE CODE
630 MEMORY &A87F
640 MC=&A880
650 TA=&A8FF
660 POKE MC,&CD
670 POKE MC+1,&60
680 POKE MC+2,&BB
690 POKE MC+3,&32
700 POKE MC+4,&FF
710 POKE MC+5,&A8
720 POKE MC+6,&C9
730 RETURN
740 REM CHECK ON SCREEN USING MACHINE CODE ROUTINE
750 CALL MC
760 DD=PEEK(TA)
770 RETURN
780 REM DEFINE LORRY AND SPIDER CHARACTERS
790 SYMBOL 250,0,&77,&22,&7E,&BE,&7E,&22,&77
800 SYMBOL 251,&24,&99,&7E,&3C,&FF,&3C,&3C,&5A
810 SYMBOL 252,0,&E0,&40,&FE,&FF,&FE,&40,&E0

```

```
820 SYMBOL 253,&BD,&42,&BD,&B5,&AD,&BD,&
42,&BD
830 RETURN
840 REM BUILD LORRY STRINGS
850 A$=""
860 FOR L=1 TO 5
870 A$=A$+CHR$(250)+STRING$(3,CHR$(206))
880 NEXT L
890 B$=""
900 FOR L=1 TO 4
910 B$=B$+CHR$(252)+STRING$(4,CHR$(206))
920 NEXT L
930 RETURN
940 REM FILL IN THE SCREEN
950 PEN 7
960 Q$=STRING$(20,CHR$(206))
970 LOCATE 1,2
980 FOR L=1 TO 24
990 PRINT Q$;
1000 NEXT L
1010 LOCATE 1,1
1020 PEN 8
1030 PRINT STRING$(9,CHR$(250));" ";STRIN
NG$(10,CHR$(252));
1040 PEN 13
1050 LOCATE 10,1
1060 PRINT CHR$(253);
1070 RETURN
1080 REM MOVE THE LORRIES
1090 PEN 9
1100 LOCATE 1,3
1110 PRINT A$
1120 LOCATE 1,7
1130 PRINT A$
1140 LOCATE 1,11
1150 PRINT A$
1160 LOCATE 1,15
1170 PRINT A$
1180 A$=RIGHT$(A$,19)+LEFT$(A$,1)
1190 PEN 7
1200 LOCATE 1,5
1210 PRINT B$
1220 LOCATE 1,9
1230 PRINT B$
1240 LOCATE 1,13
```

```
1250 PRINT B$  
1260 LOCATE 1,17  
1270 PRINT B$  
1280 B$=RIGHT$(B$,1)+LEFT$(B$,19)  
1290 RETURN
```

DOUBLE SURROUND



Here is a game you can only play with a close friend as you will have to sit in each others lap to play.

Each player will have the target of trying to surround his opponent before being surrounded himself. Sounds tricky doesn't it? Just wait until you try to play complete with some very nasty crashing and crunching noises when you finally collide.

Oh, by the way, if you happen to crash into your own trail it will mean goodbye for you so keep on your toes and off your opponents trail – and your own.

How to play

Player One uses Q and A and Z and X

Player Two uses . and , and ; and /

The program

```

10 REM DOUBLE SURROUND
20 C=3:D=3:MODE 1:BORDER 14
30 GOSUB 560
40 SYMBOL 250,&3C,&3C,&D7,&EB,&D7,&EB,&3
C,&3C
50 REM NEW WAVE START
60 X1=5:Y1=12
70 X2=35:Y2=12
80 A1=1:B1=0
90 A2=-1:B2=0
100 MODE 1
110 SOUND 1,0,4,10,0,0,15
120 IF INKEY(67)=0 THEN A1=0:B1=-1
130 IF INKEY(69)=0 THEN A1=0:B1=1
140 IF INKEY(71)=0 THEN A1=-1:B1=0
150 IF INKEY(63)=0 THEN A1=1:B1=0
160 LOCATE X1,Y1:FEN 3:PRINT CHR$(250);
170 X1=X1+A1:Y1=Y1+B1
180 XX=X1:YY=Y1
190 GOSUB 680
200 IF DD<>32 THEN GOTO 330
210 LOCATE X1,Y1:FEN 1:PRINT CHR$(250);
220 IF INKEY(39)=0 THEN A2=-1:B2=0
230 IF INKEY(31)=0 THEN A2=1:B2=0
240 IF INKEY(30)=0 THEN A2=0:B2=1
250 IF INKEY(28)=0 THEN A2=0:B2=-1
260 LOCATE X2,Y2:FEN 3:PRINT CHR$(250);
270 X2=X2+A2:Y2=Y2+B2
280 XX=X2:YY=Y2
290 GOSUB 680
300 IF DD<>32 THEN GOTO 360
310 LOCATE X2,Y2:FEN 2:PRINT CHR$(250);
320 GOTO 110
330 REM PLAYER ONE CRASHED
340 C=C-1
350 D=D+1
360 REM PLAYER TWO CRASHED
370 D=D-1
380 FOR L=10 TO 500 STEP 35
390 SOUND 1,L,30:SOUND 2,L+20,28
400 FOR Q=1 TO 20
410 INK 3,INT(RND(1)*15+5)

```

```
420 NEXT Q
430 NEXT L
440 INK 3,9
450 IF SQ(1)<>4 THEN GOTO 450
460 IF C*D<>0 THEN GOTO 50
470 REM SOMEONE HAS WON
480 LOCATE 8,12
490 PEN 1
500 PRINT"    PLAYER ";
510 IF C=0 THEN PRINT"TWO "; ELSE PRINT"
ONE ";
520 PRINT"WINS BY ";ABS(D-C);"
    ";
530 IF INKEY$<>"" THEN GOTO 530
540 IF INKEY$="" THEN GOTO 540
550 RUN
560 REM PUT MACHINE CODE INTO RAM
570 MEMORY &A97F
580 MC=&A980
590 TA=&A9FF
600 POKE MC,&CD
610 POKE MC+1,&60
620 POKE MC+2,&BB
630 POKE MC+3,&32
640 POKE MC+4,&FF
650 POKE MC+5,&A9
660 POKE MC+6,&C9
670 RETURN
680 REM READ CHARACTER FROM SCREEN
690 IF XX<1 OR XX>40 OR YY<1 OR YY>25 TH
EN DD=-1:GOTO 730
700 LOCATE XX,YY
710 CALL MC
720 DD=PEEK(TA)
730 RETURN
```

SNAKER



You are in charge of a very fast growing snake which is charging around your television screen, getting longer and longer second by second.

Your objective is to stay alive as long as you can, before your own body becomes too long and you run over yourself and self destruct.

If you run off the edge of the screen, you will just pop up on the other side and keep going, so don't try to gain any time by taking this way out.

The longer you can stay alive the higher score you will achieve but I'm afraid that the snake will get you into trouble in the end.

How to play

Guide yourself around the screen using the arrow keys to move.

You will be asked for your name and when you have typed it in and pressed ENTER the game will begin. To restart the game press the SPACE bar.

The program

```

10 REM SNAKER
20 GOSUB 470
30 HIGH=0
40 HS$=""
50 SYMBOL 250,&18,&3C,&42,&DB,&DB,&42,&3
C,&18
60 SYMBOL 251,&18,&5A,&3C,&E7,&E7,&3C,&1
8,&5A
70 REM NEW PLAYER LOOP
80 SC=0
90 MODE 1
100 LOCATE 1,10
110 INPUT "What is your name ";name$
120 IF LEN(NAME$)>10 OR NAME$="" THEN GO
TO 90
130 MODE 1
140 X=5:Y=5
150 KK=1
160 IF HIGH=0 THEN GOTO 180
170 LOCATE 1,1:PRINT "          HIGH SCO
RE OF ";HIGH;" BY ";HS$
180 REM MAIN PLAY LOOP
190 GOSUB 650
200 A=X:B=Y
210 KK=0
220 IF KK=0 THEN B=B-1:IF B=0 THEN B=25
230 IF KK=2 THEN B=B+1:IF B=26 THEN B=2
240 IF KK=1 THEN A=A+1:IF A=41 THEN A=1
250 IF KK=8 THEN A=A-1:IF A=0 THEN A=40
260 XX=A:YY=B
270 GOSUB 600
280 IF DD<>32 THEN GOTO 370
290 PEN 3
300 LOCATE X,Y:PRINT CHR$(250);
310 PEN 2
320 LOCATE A,B:PRINT CHR$(251);
330 X=A:Y=B
340 SC=SC+15

```



```

350 SOUND 1,300,5
360 GOTO 190
370 REM THE SNAKE ATE ITSELF
380 FOR Q=10 TO 450 STEP 5
390 SOUND 1,Q,4
400 SOUND 2,Q+10,3
410 INK 3,INT(RND(1)*16+4)
420 NEXT Q
430 IF SC>HIGH THEN LOCATE 1,10:PRINT"TH
E NEW HIGH SCORE":HIGH=SC:HS$=NAME$
440 IF INKEY$=" " THEN GOTO 440
450 IF INKEY$<>" " THEN GOTO 450
460 GOTO 70
470 REM PUT MACHINE CODE ABOVE RAMTOP
480 MEMORY &A97F
490 MC=&A980
500 TH=&A9
510 TL=&FF
520 POKE MC,&CD
530 POKE MC+1,&60
540 POKE MC+2,&BB
550 POKE MC+3,&32
560 POKE MC+4,TL
570 POKE MC+5,TH
580 POKE MC+6,&C9
590 RETURN
600 REM RETURN SCREEN CHARACTER
610 LOCATE XX,YY
620 CALL MC
630 DD=PEEK(&A9FF)
640 RETURN
650 Q=KK
660 FOR L=0 TO 8
670 IF INKEY(L)=0 THEN Q=L
680 NEXT L
690 IF Q<>0 AND Q<>1 AND Q<>2 AND Q<>8 T
HEN Q=KK
700 RETURN

```

THE AMSTRAD PROGRAM BOOK

Your Amstrad computer has already been hailed by the computer press as having 'graphics and sound facilities superior to the Spectrum and Oric'.

To meet the demands of the Amstrad we have produced a wide ranging selection of programs to make full use of the colour, sound and speed available to you.

Here are arcade style games to test your reaction skills and mental agility as you battle with aliens from outer space.

Adventures to make you face the dragons and dungeons to save the princess.

Brainteasers to make you skip through minefields and worse.

Programmers aids to help you build up your own computing skills.

This is the AMSTRAD PROGRAM BOOK with something for everyone.

THE AUTHOR

Peter Goode is the author of several best selling program books including the Atari 600XL Program Book and Memotech MTX Program Book.



5 2284

30.50

25.95

ISBN 0-9465-7624-6



90000

9 780946 576241